

Министерство образования и науки Российской Федерации
Федеральное государственное бюджетное образовательное
учреждение высшего профессионального образования
**ТОМСКИЙ ГОСУДАРСТВЕННЫЙ УНИВЕРСИТЕТ СИСТЕМ
УПРАВЛЕНИЯ И РАДИОЭЛЕКТРОНИКИ**
Кафедра автоматизированных систем управления (АСУ)

М. Ю. Катаев

**ОБРАБОТКА
ЭКСПЕРИМЕНТАЛЬНЫХ
ДАННЫХ НА ЭВМ**

**Методические указания
по выполнению лабораторной работы
в среде Scilab**

2015

Корректор: Осипова Е. А.

Катаев М. Ю.

Обработка экспериментальных данных на ЭВМ: методические указания по выполнению лабораторной работы в среде Scilab. — Томск: Факультет дистанционного обучения, ТУСУР, 2015. — 116 с.

© Катаев М. Ю., 2015
© Факультет дистанционного
обучения, ТУСУР, 2015

ОГЛАВЛЕНИЕ

ВВЕДЕНИЕ.....	6
1 УСТАНОВКА SCILAB	8
2 ОСНОВЫ РАБОТЫ В SCILAB	23
2.1 Консоль Scilab	23
2.2 Встроенный редактор скриптов.....	25
2.3 Переключение панелей	27
2.4 Использование команды exes	27
2.5 Пакетная обработка	29
3 ОСНОВНЫЕ ЭЛЕМЕНТЫ ЯЗЫКА SCILAB	32
3.1 Определение вещественных переменных	32
3.2 Имена переменных	34
3.3 Комментарии и продолжение строки	35
3.4 Элементарные математические функции.....	36
3.5 Предопределенные математические константы	36
3.6 Логический тип.....	37
3.7 Комплексные числа	38
3.8 Целые числа	39
3.8.1 Обзор целых чисел	39
3.8.2 Преобразование целых типов	41
3.8.3 Выход за пределы диапазона и проблемы переносимости	42
3.9 Целые числа и числа с плавающей точкой	42
3.10 Переменная ans	44
3.11 Строки	45
3.12 Динамическая типизация переменных.....	45
4 МАТРИЦЫ.....	47
4.1 Обзор.....	47
4.2 Создание вещественных матриц.....	48

4.3	Пустая матрица []	50
4.4	Определение размера матрицы.....	50
4.5	Работа с элементами матрицы	52
4.6	Оператор ":"	53
4.7	Генерация единичной матрицы. Функция eye	57
4.8	Динамическое изменение размера матрицы	58
4.9	Оператор "\$"	59
4.10	Арифметические операции	61
4.11	Поэлементные операции	63
4.12	Эрмитово сопряжение и транспонирование	64
4.13	Умножение векторов.....	65
4.14	Сравнение вещественных матриц.....	66
4.15	Числа с плавающей точкой в качестве индексов.....	68
4.16	Элементарные функции	70
5	ОПЕРАТОРЫ ВЕТВЛЕНИЯ И ЦИКЛА	73
5.1	Оператор if.....	73
5.2	Оператор select	75
5.3	Оператор for.....	77
5.4	Оператор while.....	80
5.5	Инструкции break и continue	81
6	ФУНКЦИИ.....	83
6.1	Обзор.....	83
6.2	Создание собственной функции	86
6.3	Библиотеки функций.....	88
6.4	Управление выходными переменными.....	92
6.5	Уровни стека вызовов	93
6.6	Инструкция return.....	95
6.7	Отладка функций. Инструкция pause	96

7 ПОСТРОЕНИЕ ГРАФИКОВ	100
7.1 Обзор графических возможностей Scilab.....	100
7.2 Отображение двумерных графиков.....	101
7.3 Контурные графики.....	102
7.4 Подписи на графиках	107
7.5 Экспорт изображений	110
8 ЛАБОРАТОРНАЯ РАБОТА	111
8.1 Задание на лабораторную работу	111
8.2 Выбор варианта лабораторной работы.....	112
8.3 Варианты заданий.....	112

ВВЕДЕНИЕ

За последние десять лет разработано множество вариантов программных средств для решения задач моделирования. Среди этих программных продуктов встречаются бесплатные и достаточно дорогие. Свободно распространяемый пакет Scilab был разработан в 1994 году во Франции, в Национальном исследовательском институте информатики и автоматизации INRIA и Национальной школе дорожного ведомства ENPC. С 2003 года поддержкой Scilab занимается консорциум Scilab Consortium.

В рамках пакета Scilab присутствует пакет Scicos (Scilab Connected Object Simulator) и обеспечивает возможность визуального моделирования динамических систем. Эти моделируемые системы могут быть как непрерывными, так и дискретными. Scicos имеет дружественный графический интерфейс пользователя для редактирования моделей, состоящих из соединенных блоков Scicos, которые могут быть найдены в палитрах Scicos или определены пользователем.

В состав пакета Scilab входит большая библиотека математических функций, расширяемая программами, написанными на языках высокого уровня, например на C или Fortran 77. В состав пакета входит интегрированный язык высокого уровня, отличающийся в некоторой степени от языка C.

Scilab — пакет числовой обработки данных. Он очень хорош для обработки сигналов. Основной режим работы — командный. К основным функциям пакета Scilab можно отнести работу с матрицами (в том числе разреженными), решение обыкновенных дифференциальных уравнений, численное дифференцирование и интегрирование, построение двумерных и трехмерных графиков по формулам и по результатам расчетов, решение задач линейного программирования, а также возможность создания пользовательских программ. Пакет является кроссплатформенным, на сайте проекта (<http://www.scilab.org/>) можно найти самые последние сборки для

различных операционных систем. На русском языке подробную информацию и пособия по работе с пакетом можно получить как на сайте http://www.csa.ru/~zebra/my_scilab/, так и на сайте <http://www.scilab.land.ru/>.

Наличие доступа к функциям Scilab при создании моделей очень важно.

1. Пользователь Scicos часто имеет необходимость использовать функции Scilab, например предназначенные для создания фильтров обработки сигналов и др.

2. Язык программирования Scilab может использоваться для пакетной обработки множества задач моделирования, или, обобщая — модели, разработанные Scicos, могут использоваться как функции в Scilab.

3. Графические возможности Scilab могут использоваться для последующей обработки результатов моделирования.

Есть и недостаток у Scilab — ввиду того, что пакет имеет интерпретируемый язык, для очень больших моделей некоторые функции редактора могут быть медленными; ограниченные возможности графического интерфейса.

Отметим, что в основе пособия лежит перевод «Введение в Scilab» Michael Baudin.

1 УСТАНОВКА SCILAB

В этом разделе мы рассмотрим способ получения и установки дистрибутивов Scilab, основные компоненты, поставляемые в составе дистрибутива.

Программный пакет Scilab объединяет в себе развитый язык программирования и обширную библиотеку численных алгоритмов, охватывающую многие области научных и технических вычислений. Язык программирования Scilab относится к числу интерпретируемых языков высокого уровня, предоставляя пользователю возможность напрямую манипулировать математическими конструкциями, такими как матрицы или полиномы. Тем самым достигается большая скорость и простота написания программ. Язык Scilab допускает расширение посредством определения пользовательских типов данных. При этом стандартным операциям, например арифметическим операторам или операторам сравнения, можно придать особый смысл применительно к пользовательским типам данных. Пользователи пакета могут разрабатывать собственные модули расширения для решения конкретных задач. Возможен также вызов из Scilab функций, реализованных на других языках программирования, в частности на Fortran или C, благодаря чему сторонние библиотеки могут быть использованы, как если бы они были частью встроенных средств пакета.

В соответствии с принципами свободного программного обеспечения, Scilab распространяется бесплатно на основе лицензии Cecill. Дистрибутив Scilab включает исходный код, поэтому заинтересованный пользователь может самостоятельно исследовать внутреннее устройство пакета и особенности его работы. Скомпилированные версии пакета Scilab доступны для операционных систем Windows, Linux и Mac OS. Справочная документация переведена на многие языки мира. Scilab предоставляет чрезвычайно богатый набор программных средств для научных и инженерных

расчетов. Функциональные возможности Scilab охватывают большинство разделов научных вычислений, включая:

- линейную алгебру и разреженные матрицы,
- полиномы и рациональные функции,
- интерполяцию и аппроксимацию,
- линейную, квадратичную и нелинейную оптимизацию,
- обыкновенные дифференциальные уравнения, дифференциально-алгебраические уравнения,
- классическое и робастное управление, решение линейных матричных неравенств,
- оптимизацию дифференцируемых и недифференцируемых функций,
- обработку сигналов,
- математическую статистику.

Версия Scilab для ОС Windows содержит программу «установщик», которая значительно облегчает установку и настройку пакета. В ходе установки на рабочий стол добавляется ярлык приложения, при выборе которого отображается консоль Scilab. Для платформ на базе процессоров Intel в Scilab используется библиотека Intel Math Kernel Library (MKL), позволяющая существенно повысить скорость вычислений.

Кроме того, Scilab содержит значительное число функций для построения графиков, а также мощное средство визуального моделирования Xcos, которое объединяет в себе возможности редактора моделей и симулятора.

Итак, чтобы получить пакет программ Scilab, надо зайти на сайт: <http://www.scilab.org> (см. рис. 1.1).

Для ознакомления с содержимым сайта и пакетом программ можно обратиться к разделу: <http://www.scilab.org/scilab/about> (см. рис. 1.2). Отметим, что сайт является англоязычным, и при отсутствии знаний английского языка, можно воспользоваться переводчиками English-Русский (например, <https://translate.google.ru>).




Рис. 1.1 — Главное окно сайта Scilab




Рис. 1.2 — Информация о возможностях Scilab

Для получения программы (иначе, скачивания), необходимо перейти на вкладку download: <http://www.scilab.org/download/5.5.1> (см. рис. 1.3).




Рис. 1.3 — Окно для получения программы Scilab

Из рисунка 1.3 видно, что для операционной системы Windows можно получить либо 32-битную, либо 64-битную версию программы Scilab. Для получения необходимо вызвать ту или иную опцию, например Scilab 5.5.1 — 32-bit Windows.

После получения программы необходимо переходить к ее установке на компьютере. После вызова программы появится окно выбора языка установки (см. рис. 1.4). Программа установки проверяет типичный выбор пользователя и автоматически выбирает язык установки (для России это «Русский»), который легко может быть заменен.




Рис. 1.4 — Окно выбора языка установки

После выбора языка установки появляется окно установки Scilab (см. рис. 1.5).




Рис. 1.5 — Окно установки Scilab

При нажатии на кнопку «Далее», появляется окно принятия Лицензионного соглашения при установке (см. рис. 1.6). Необходимо выбрать пункт «Я принимаю условия соглашения» и после этого выбрать пункт «Далее».




Рис. 1.6 — Окно принятия Лицензионного соглашения при установке

Следующий этап установки связан с выбором директории установки (см. рис. 1.7). Автоматически программа выбирает директорию установки «Program Files», однако пользователь может выбрать произвольную директорию установки и диск.





Рис. 1.7 — Окно выбора директории установки

После выбора опции «Далее» происходит переход в окно выбора компонент (см. рис. 1.8 а,б). Автоматически выбраны практически все компоненты. Пользователь вправе выбрать произвольный вариант компонент. Надо отметить, что выбор компонент может повлиять на работоспособность пакета программ в целом.



а

б

Рис. 1.8 — Окно выбора компонент Scilab

Отметим, что о каждом компоненте программы можно прочитать краткую информацию в разделе «About modules». После выбора компонент необходимо выбрать опцию «Далее», после чего начинается установка программы на компьютер (см. рис. 1.9).




Рис. 1.9 — Окно установки программы Scilab

Длительность установки зависит от типа и производительности компьютера. После завершения установки появляется окно, представленное на рисунке 1.10. Пользователь может открыть сразу программу, если воспользуется опцией «Launch Scilab», или открыть через выбор программы на рабочем столе или в меню «Пуск».




Рис. 1.10 — Завершающее окно установки программы

После запуска программы появляется рабочее окно, которое показано на рисунке 1.11. Далее необходимо изучить основные элементы программы, связанные с меню. На рисунке 1.12 показано содержание меню «Файл». Содержание этого меню является типичным для большинства программ.




Рис. 1.11 — Основное рабочее окно программы Scilab

После запуска Scilab на экране появится основное окно приложения (см. рис. 1.11). Окно содержит меню, панель инструментов и рабочую область. Признаком того, что система готова к выполнению команды, является наличие знака приглашения -->, после которого расположен активный курсор. Рабочую область со знаком приглашения обычно называют командной строкой. Ввод команд в Scilab осуществляется с клавиатуры. Нажатие клавиши Enter заставляет систему выполнить команду и вывести результат.

Понятно, что все выполняемые команды не могут одновременно находиться в поле зрения пользователя. Поэтому просмотреть ту информацию, которая покинула видимую часть окна, можно, если воспользоваться

стандартными средствами просмотра информации в окне Windows, например полосами прокрутки или клавишами перемещения курсора Page Up, Page Down. Клавиши «Стрелка вверх» и «Стрелка вниз» так же управляют курсором, однако в Scilab они имеют другое назначение. Эти клавиши позволяют вернуть в командную строку ранее введенные команды или другую входную информацию, так как вся эта информация сохраняется в специальной области памяти. Так, если в пустой активной командной строке нажать клавишу (стрелка вверх), то появится последняя вводимая команда, повторное нажатие вызовет предпоследнюю и так далее. Клавиша (стрелка вниз) выводит команды в обратном порядке. Таким образом, можно сказать, что вся информация в рабочей области находится или в зоне просмотра или в зоне редактирования.

Важно знать, что в зоне просмотра нельзя ничего исправить или ввести. Единственная допустимая операция — это выделение информации с помощью мыши и копирование ее в буфер обмена, например для дальнейшего помещения в командную строку. Зона редактирования это фактически командная строка. В ней действуют элементарные приемы редактирования:

- стрелка вправо — перемещение курсора вправо на один символ;
- стрелка влево — перемещение курсора влево на один символ;
- Home — перемещение курсора в начало строки;
- End — перемещение курсора в конец строки;
- Del — удаление символа после курсора;
- Backspace — удаление символа перед курсором.

Кроме того, существуют особенности ввода команд. Если команда заканчивается точкой с запятой «;», то результат ее действия не отображается в командной строке. В противном случае, при отсутствии знака «;», результат действия команды сразу же выводится в рабочую область.




Рис. 1.12 — Содержание меню «Файл»

На рисунке 1.13 показано содержание меню «Правка», в котором представлены типичные для офисных программ опции.




Рис. 1.13 — Содержание меню «Правка»

Важной является опция меню «Управление», которое показано на рисунке 1.14. В этом меню присутствуют опции запуска и возобновления программы, завершения и приостановления работы программы.




Рис. 1.14 — Содержание меню «Управление»

На рисунке 1.15 показано содержание меню «Инструменты».




Рис. 1.15 — Содержание меню «Инструменты»

В состав программы также входит компонент Xcos (или Scicos) — инструмент для редактирования блочных диаграмм и симуляции (аналог Simulink в пакете MATLAB). Программные компоненты ATOMS (Automatic mOdules Management for Scilab) [<https://atoms.scilab.org/>] служат для управления внешними модулями.

Справочная информация приведена в элементе меню «Справка» (см. рис. 1.16).




Рис. 1.16 — Содержание меню «Справка».

На рисунке 1.17 приведено окно справочной системы программы Scilab. В левой части экрана приведены элементы меню, а в правой (большой) части экрана описание вызываемого пользователем материала.




Рис. 1.17 — Окно справочной системы Scilab

Наиболее простым способом получить справку по возможностям пакета Scilab является функция «help». Окно справки Scilab показано на рисунке 1.17. Для его отображения наберите help в консоли и нажмите клавишу <Enter>:

```
-->help
```

Если вас интересует информация о конкретной функции (например, optim), вы можете пролистать содержание справки, найти раздел, посвященный оптимизации, и выбрать optim, после чего будет отображено окно со справкой по данной функции. Однако более удобным способом получить информацию о конкретной функции, если вы знаете ее название, является использование команды help с указанием имени интересующей функции:

```
-->help optim
```

В этом случае Scilab автоматически отобразит окно справки, содержащее информацию о выбранной функции. Если функции с указанным именем не существует, будет отображено сообщение об ошибке. Вызов раздела справки «основные функции» приводит к появлению окна вида, представленного на рисунке 1.18, а более детальной информации о какой-либо функции — к появлению окна на рисунке 1.19.




Рис. 1.18 — Окно справки Scilab. Основные функции




Рис. 1.19 — Детальный просмотр справки функции «acos»

В программе Scilab для простоты и быстроты обучения приведено много примеров в разделе меню «Справка» — «Примеры». После выбора этого элемента меню появляется окно, которое изображено на рисунке 1.20.




Рис. 1.20 — Окно примеров Scilab

Некоторые примеры являются графическими, другие предполагают пошаговое выполнение, ожидая нажатия клавиши <Enter> для перехода к следующему шагу. Сопутствующие этим примерам скрипты размещаются в соответствующих подкаталогах каталога Scilab.

В программе Scilab можно выполнить и элементы имитационного моделирования в компоненте Xcos (или Scicos). Перечислим основные составные части Scicos.

– Графический редактор Scicos предоставляется иерархическому графическому редактору для строительства моделей динамических систем, с использованием блок-схемы. Многие predetermined blocks are also provided in various palettes. New blocks can be defined by the user in C, Fortran or Scilab.

– Компилятор: компилятор Scicos uses a template description, usually generated by the Scicos editor, for the construction of planning tables.

которые могут тогда использоваться тренажером и кодовой функцией поколения.

– Симулятор: симулятор Scicos использует таблицы планирования и другую информацию, предоставленную компилятором, чтобы управлять моделированиями. Симулятор имеет гибридную природу, при которой он должен иметь дело с дискретными и непрерывными системами времени и событиями.

Основные окна Xcos приведены на рисунке 1.21.




Рис. 1.21 — Окно компоненты Xcos

Работа в этой компоненте не входит в компетенции при выполнении лабораторных и поэтому не будет рассматриваться детально.

Программа Scilab имеет не только внутреннее описание в файлах, но и в Интернете. Wiki-энциклопедия Scilab, представляющая открытый источник для обмена опытом между пользователями и разработчиками пакета, располагается по адресу <http://wiki.scilab.org>.

Программисты смогут здесь найти пошаговые инструкции для компиляции Scilab, информацию о библиотеках, необходимых для работы различных версий пакета, указания по работе с хранилищем исходного кода и т. д.

2 ОСНОВЫ РАБОТЫ В SCILAB

В этом разделе мы сделаем наши первые шаги в Scilab и рассмотрим различные пути работы с пакетом:

- 1) используя консоль Scilab в режиме диалога,
- 2) используя функцию `exec` для выполнения предварительно написанных алгоритмов,
- 3) используя терминал операционной системы и возможности пакетной обработки.

2.1 Консоль Scilab

Простейшим способом использования Scilab является непосредственный ввод команд в консоли. Результат выполнения команды при этом отображается сразу же после ее ввода и выполнения. Все примеры в настоящем руководстве можно выполнить, копируя соответствующие команды в консоль, так что читатель может самостоятельно экспериментировать с рассматриваемыми возможностями Scilab. Пошаговое выполнение является наиболее эффективным способом для того, чтобы понять поведение готовых программ, и чаще всего позволяет достаточно быстро перейти к разработке собственных алгоритмов.

В следующем примере мы воспользуемся функцией `disp` для отображения строки " ПРИВЕТ!":

```
-->s = "ПРИВЕТ!"
s =
ПРИВЕТ!
-->disp (s)
ПРИВЕТ!
```

Символ "-->" представляет собой приглашение Scilab и отображается автоматически, когда Scilab ожидает ввода очередной команды от пользователя. Набрав инструкцию `s=" ПРИВЕТ!"` и нажав клавишу <Enter>, мы укажем Scilab выполнить команду: создать переменную `s`, содержащую заданный текст.




Рис. 2.1 — Подсказка отображается при нажатии клавиши <Tab> после набора оператора disp

Реакция Scilab на ввод команды подтверждает, что переменная была создана и ее значение установлено. Теперь, набрав `disp(s)`, можно отобразить содержимое созданной переменной. Для ввода команд используется клавиатура, точно так же как в обычном текстовом редакторе. Для перемещения курсора в пределах строки используются клавиши `←` и `→`. Для удаления неверно введенного символа используются кнопки `<Backspace>` и `<Suppr>`. Клавиши `"` и `#` позволяют перемещаться по истории ранее введенных команд, в том числе в рамках предыдущих сеансов работы с пакетом.

При нажатии на клавишу `<Tab>` отображается подсказка (рис. 2.1), где перечислены команды, имена которых начинаются с введенных символов (в данном случае `disp`). С помощью клавиш `"`, `#` и `<Enter>` можно выбрать из списка необходимую команду. Подсказки отображаются для имен функций, переменных, файлов и графических дескрипторов, тем самым ускоряя и упрощая работу со Scilab.

2.2 Встроенный редактор скриптов

Реализованный ранее и сохраненный в файл алгоритм, представляющий последовательность инструкций для решения некоторой задачи, называется скриптом. Версия Scilab 5 предлагает новый редактор, упрощающий разработку собственных скриптов. Внешний вид окна редактора показан на рисунке 2.2. В данном случае пользователь редактирует скрипт, содержащий команды рассмотренного выше примера, где в консоль выводится строка "Hello World!". Запустить редактор можно из меню Инструменты > Текстовый редактор (Applications > Editor) либо из консоли, набрав команду `--> editor ()`




Рис. 2.2 — Окно редактора скриптов

Редактор позволяет работать с несколькими открытыми файлами одновременно. Наиболее часто используемые команды редактора располагаются в меню Выполнение (Execute):

- Загрузить в Scilab (Load into Scilab) — позволяет выполнить все команды некоторого скрипта так, как будто эти команды последовательно

вводятся в консоли. При этом результат выполнения инструкций, оканчивающихся символом ”;”, не отображается.

– Вычислить выделенное (Evaluate Selection) — позволяет выполнить выделенные инструкции.

– Выполнить файл в Scilab (Execute File Into Scilab) — загружает на исполнение файл подобно тому, как это делается с использованием функции `exec` (см. раздел 2.4). При этом в консоль будут выводиться лишь результаты работы печатающих функций, например `disp`.

Меню Правка (Edit) предлагает полезную возможность автоматического форматирования отступов Исправить отступы (Correct Indentation). Эта возможность позволяет структурировать текст программы, что существенно упрощает чтение блоков в таких конструкциях, как `if`, `for` и т. д. Выделив несколько строк и нажав правую кнопку мыши (или комбинацию `Control+Click` в Mac OS), можно отобразить контекстное меню, представленное на рисунке 2.3. Контекстное меню содержит ряд полезных команд:




Рис. 2.3 — Контекстное меню в редакторе скриптов

– Вычислить выделенное (Execute selection in Scilab) — выполнить выделенные команды.

– Править выделенное в новой вкладке (Edit selection in a new tab) — открыть новую вкладку и скопировать туда выделенный фрагмент.

– Справка по '...' (Help about '...') — отобразить страницу помощи, связанную с выделенной командой.

2.3 Переключение панелей

Предположим, что в какой-то момент открыты окно консоли и редактора. Переключение между двумя окнами может представлять неудобство, поэтому реализована возможность разместить окно редактора в качестве панели в пределах основного окна Scilab. Все окна в Scilab, включая консоль, текстовый редактор, окно справки и окна вывода графических результатов, могут быть сгруппированы подобным образом. Могут быть ситуации, когда совместно размещены четыре таких окна. Для того чтобы поместить некоторое окно рядом с другим, необходимо перетянуть его в область, занимаемую вторым окном. Для этого следует нажать левой клавишей мыши на заголовок перемещаемого окна и, удерживая клавишу мыши, переместить указатель в область, где должна размещаться панель. При этом серыми пунктирными линиями будет отображено предлагаемое положение перетаскиваемого окна (один из четырех вариантов: слева, справа, снизу или сверху). Выбрав один из вариантов, отпустите левую клавишу мыши, после чего панель займет свое положение. Если отпустить клавишу мыши над центром области, панель будет добавлена в качестве вкладки.

2.4 Использование команды `exec`

Последовательность из нескольких команд Scilab, предназначенную для многократного выполнения, удобно записать в файл, используя редактор скриптов Scilab. Для того чтобы впоследствии запустить данный

скрипт на выполнение из консоли, применяется команда `exec`, после которой указывается имя файла. В зависимости от содержимого файла расширением будет либо `.sce`, либо `.sci`:

– файлы с расширением `.sci` содержат определения функций Scilab — выполнение таких файлов приводит к загрузке данных функций в рабочее окружение пакета, после чего их можно использовать точно так же, как встроенные функции Scilab;

– файлы с расширением `.sce` могут содержать как определения функций, так и исполняемые инструкции; обычно в результате выполнения такого файла вычисляются значения переменных, происходит вывод результатов в консоль, отображение графиков, чтение или запись файлов и т. д.

Представим, что файл `myscript.sce` содержит всего одну строку:

```
disp ("ПРИВЕТ!")
```

Используя функцию `exec` в консоли, мы запускаем данный скрипт на выполнение:

-e инструкция выполнить инструкцию

-f файл выполнить скрипт, записанный в файле

-l язык выбрать язык интерфейса: `fr` для французского, `en` для английского, `ru` для русского. По умолчанию языком интерфейса является английский. Для установки языка по умолчанию, используемого при загрузке Scilab без задания параметра `-l`, применяется команда `setdefaultlanguage`. После выполнения этой команды программу необходимо перезапустить.

Команда `getlanguage` позволяет узнать, какой язык используется программой в данный момент.

```
-->exec ("myscript . sce ")
-->disp ("ПРИВЕТ!")
ПРИВЕТ!
```

На практике при отладке сложных алгоритмов в интерактивном режиме чаще всего используется комбинация вызовов функций `exec` и `disp`.

Таблица 2.1 — Параметры командной строки Scilab

<code>-mem N</code>	установить начальный размер стека равным N байт
<code>-ns</code>	отключить выполнение скрипта <code>scilab.start</code> при загрузке Scilab
<code>-nb</code>	отключить вывод приветствия Scilab
<code>-pouserstartup</code>	отключить выполнение пользовательских стартовых скриптов <code>SCIHOME/.scilab</code> (в Linux и Mac OS) или <code>SCIHOME/scilab.ini</code> (в Windows)
<code>-nw</code>	запустить Scilab в режиме командной строки с подключением дополнительных возможностей (например отображения графиков)
<code>-nwni</code>	запустить Scilab в режиме командной строки без дополнительных возможностей
<code>-version</code>	отобразить текущую версию Scilab и завершить работу

2.5 Пакетная обработка

Иным способом работы с пакетом Scilab является его запуск из командной строки операционной системы без отображения оконного интерфейса. Для этого используются параметры командной строки, приведенные в таблице 2.1. Вне зависимости от используемой операционной системы исполняемые файлы размещаются в каталоге `scilab-<version>/bin`, где `<version>` представляет номер установленной на компьютере версии Scilab (например, '5.2.1'). Аргументы командной строки должны следовать за именем исполняемого файла, как будет показано ниже. Параметр `-nw` позволяет отключить отображение окна консоли, при этом ввод команд и вывод результатов будет происходить в текстовом режиме в терминале операционной системы. Опция `-nwni` позволяет запустить Scilab без графических библиотек. В этом случае функции отображения (например, вывод графиков) будут заблокированы, а попытка их использовать приведет к ошибке.

Для удобства пользователя в подкаталоге `bin` основного каталога Scilab присутствуют файлы, предназначенные для запуска Scilab в различных режимах без использования параметров командной строки.

Для операционных систем семейства Windows в дистрибутивах Scilab поставляются два исполняемых файла. Первый из них, WScilex.exe, запускает интерактивную графическую (оконную) консоль Scilab, описанную выше. Именно на этот файл указывает ярлык, создаваемый программой-установщиком Scilab на рабочем столе Windows. Вторым исполнимым файлом, Scilex.exe, служит для запуска консоли в текстовом режиме, использующем стандартные средства терминала Windows, как при использовании опции -nw. Указав при запуске Scilex.exe параметр -nwni, можно полностью отключить загрузку графических библиотек, после чего вывод диаграмм и другие подобные возможности станут недоступны.

Наиболее полезным параметром командной строки является -f, позволяющий выполнить команды, содержащиеся в некотором файле. Такой метод называется пакетной обработкой. Предположим, что файл myscript2.sce содержит следующие строки (команда quit применяется для выхода из Scilab):

```
disp ("ПРИВЕТ!")
quit ()
```

По умолчанию в интерактивном режиме Scilab ожидает от пользователя ввода следующей команды. Команда quit используется для завершения сессии и выхода из Scilab. Для дальнейшей демонстрации предположим, что мы создали на диске C: каталог scripts и поместили в него файл скрипта myscript2.sce.

Любая строка, начинающаяся символами " //", считается комментарием и игнорируется интерпретатором Scilab. Для того чтобы проверить, что по умолчанию Scilab остается в интерактивном режиме, мы исключим команду quit, используя символы комментария:

```
disp ("ПРИВЕТ!")
//quit()
```

Если теперь набрать в терминале операционной системы инструкцию `scilex -f myscript2.sce`, Scilab отобразит строку "ПРИВЕТ!" и перейдет в режим ожидания. Теперь, чтобы выйти, необходимо ввести команду `quit` вручную.

3 ОСНОВНЫЕ ЭЛЕМЕНТЫ ЯЗЫКА SCILAB

Как уже отмечалось, Scilab относится к числу языков программирования высокого уровня и позволяет эффективно манипулировать сложными структурами данных. В данной главе будут рассмотрены основные возможности языка, а именно создание вещественных матриц, использование элементарных математических функций и т. д. Если бы Scilab ограничивался только этими возможностями, он был бы не более чем продвинутым настольным калькулятором. Конечно, возможности Scilab намного шире, и в последующих разделах мы рассмотрим работу с другими типами данных (логическими переменными, комплексными и целыми числами, а также строками).

Следует с самого начала усвоить, что большинство объектов в Scilab являются матрицами. Все вещественные, комплексные, целочисленные и логические переменные, строки и полиномы представлены в виде матриц. Не являются матрицами списки и другие составные структуры — эти типы данных в настоящем руководстве не рассматриваются. Матрица представляет набор элементов одного типа в виде таблицы, содержащей некоторое число строк и столбцов. Тип элементов определяет набор операций, в которых матрица может участвовать. В данной главе мы рассмотрим работу с матрицами различных типов, начиная с наиболее часто используемого типа — вещественной матрицы.

3.1 Определение вещественных переменных

Scilab предоставляет возможности для работы как с вещественными, так и с комплексными числами. Это может приводить к путанице, если не вполне ясен контекст. Комплексные переменные будут рассмотрены далее по тексту, как обобщение вещественных переменных. В большинстве случаев вещественные и комплексные переменные ведут себя одинаково, хотя

иногда обработка комплексных величин требует отдельного внимания. Для простоты будем далее рассматривать вещественные переменные, делая необходимые оговорки в том случае, если поведение комплексных переменных имеет особенности применительно к обсуждаемому вопросу.

В данном разделе мы научимся создавать вещественные переменные и выполнять с ними простые манипуляции. В Scilab, поскольку он является интерпретируемым языком, нет необходимости объявлять переменную до ее использования. Переменная создается в тот момент, когда ей впервые присваивается значение.

В следующем примере мы создаем переменную x , которой присваиваем значение 1, после чего выполняем умножение на 2. Оператор "=" в Scilab используется для установки значения переменной (в отличие от оператора "==", который применяется для проверки на равенство).

```
-->x = 1
x =
1.

-->x = x * 2
x =
2.
```

Таблица 3.1 — Элементарные математические операторы Scilab

№	Математический оператор	Функция
1	+	сложение
2	-	вычитание
3	*	умножение
4	/	деление справа, т. е. $x/y = xy^{-1}$
5	\	деление слева, т. е. $x\backslash y = x^{-1}y$
6	^	возведение в степень, т. е. x^y
7	**	возведение в степень (эквивалентно ^)
8	'	эрмитово сопряжение (комплексное сопряжение и транспонирование)

Значение переменной отображается после выполнения каждой инструкции. Если это нежелательно, после инструкции ставится символ ";", как показано в следующем фрагменте:

```
-->y = 1;
-->y = y * 2;
```

Все привычные алгебраические операторы доступны в Scilab (см. табл. 3.1). Стоит отметить, что оператор возведения в степень представляется символом карет '^', поэтому вычисление x^2 в Scilab выполняется посредством выражения x^2 или эквивалентного ему выражения $x**2$. Оператор эрмитова сопряжения '''' будет более подробно рассмотрен далее, где мы коснемся работы с комплексными числами.

3.2 Имена переменных

Имена переменных в Scilab могут иметь произвольную длину, однако лишь первые 24 символа имени являются значимыми, поэтому во избежание ошибок следует использовать имена длиной до 24 символов. Допустимыми символами в именах переменных являются латинские буквы, цифры, а также символы '%', '_', '#', '!', '\$', '?'. Следует отметить, что некоторые переменные, имена которых начинаются символом '%', имеют особый смысл в Scilab: как будет показано в разделе 3.5, такие переменные являются предопределенными математическими константами. Scilab чувствителен к регистру символов, поэтому в следующем примере интерпретатор считает переменные A и a различными:

```
-->A = 2
A =
2.
-->a = 1
a =
1.
-->A
A =
2.
-->a
a =
1.
```

Таблица 3.2 — Элементарные математические функции: тригонометрические

acos acosd acosm acosm acot acotd acoth acsc acscd acsch asec asecd asech asin asind
asinh asinhm asinm atan atand atanh atanhm atanm cos cosd cosh coshm cosm cotd cotg coth
cothm csc cscd csch sec secd sech sin sinc sind sinh sinhm sinm tan tand tanh tanhm tanm

Таблица 3.3 — Элементарные математические функции: прочие

exp expm log log10 log1p log2 logm max maxi min mini
modulo pmodulo sign signm sqrt sqrtm

3.3 Комментарии и продолжение строки

Для оформления комментариев в Scilab применяется синтаксис, заимствованный из языка C++: строки, начинающиеся двумя последовательными символами косой черты "//", считаются комментариями и игнорируются при выполнении. В отличие от C и C++, в Scilab отсутствуют многострочные комментарии (ограниченные символами "/*" и "*/").

Если инструкция слишком длинна и не помещается в одну строку, ее можно записать в несколько строк с использованием символа продолжения строки (две точки ".."). Всякая строка, завершающаяся двумя последовательными точками, считается продолженной на следующую строку. Вычисление выражения, записанного в несколько строк, происходит после ввода последней строки.

В следующем фрагменте кода приводится пример комментария и продолжения строки:

```
-->// Это комментарий.
-->x = 1..
-->+ 2..
-->+ 3..
-->+ 4
x =
10.
```

3.4 Элементарные математические функции

В табл. 3.2 и 3.3 представлен список элементарных математических функций, доступных в Scilab. Большинство функций в этом списке принимают один аргумент на вход и возвращают единственное значение.

Таблица 3.4 — Предопределенные математические константы

%i	мнимая единица i
%e	основание натурального логарифма e
%pi	число π

Все элементарные функции векторизованы в том смысле, что могут быть применены к матрицам, при этом в результате также получаем матрицы. Это свойство позволяет обрабатывать данные более эффективно, без использования циклов.

В следующем примере мы воспользуемся стандартными функциями \sin и \cos для проверки равенства $\cos^2 x + \sin^2 x = 1$:

```
-->x = cos (2)
x =
- 0.4161468
-->y = sin (2)
y =
0.9092974
-->x ^ 2 + y ^ 2
ans =
1.
```

3.5 Предопределенные математические константы

В Scilab имена математических констант начинаются символом "%". Примеры приведены в табл. 3.4. В следующем примере переменная %pi используется для проверки равенства $\cos^2 x + \sin^2 x = 1$:

```
-->c = cos ( %pi )
c =
- 1.
```

```

-->s = sin ( %pi )
s =
1.225D -16
-->c ^ 2 + s ^ 2
ans =
1.

```

Тот факт, что вычисленное значение $\sin(\pi)$ не равно в точности 0, является следствием ограниченной разрядности двоичного представления чисел с плавающей точкой в памяти компьютера.

3.6 Логический тип

Переменная логического типа может хранить значения истина или ложь. В Scilab истина представляется литералами %t или %T, а ложь — %f или %F. В табл. 3.5 перечислены логические операторы и операторы сравнения, которые используются в пакете. Операторы сравнения принимают на вход данные любого из основных типов (вещественные, комплексные и целые числа, строки) и возвращают логическое значение. Операторы сравнения также рассматриваются далее по тексту, посвященному сравнению матриц.

Таблица 3.5 — Логические операторы и операторы сравнения

a & b	логическое 'И' (конъюнкция)
a b	логическое 'ИЛИ' (дизъюнкция)
!a	логическое отрицание
a == b	истина, если a равно b
a ~= b	или a <> b истина, если a и b различаются
a < b	истина, если a меньше b
a > b	истина, если a больше b
a <= b	истина, если a меньше либо равно b
a >= b	истина, если a больше либо равно b

Таблица 3.6 — Функции Scilab для работы с комплексными числами

real	возвращает действительную часть комплексного числа
imag	возвращает мнимую часть комплексного числа
imult	умножает число на мнимую единицу
isreal	проверяет отсутствие мнимой части

Следующий пример иллюстрирует выполнение операций с логическими типами:

```
-->a = %T
a =
T
-->b = ( 0 == 1 )
b =
F
-->a & b
ans =
F
```

3.7 Комплексные числа

Комплексные числа в Scilab представляются в виде пары вещественных чисел. Предопределенная константа %i содержит значение мнимой единицы i , удовлетворяющее равенству $i^2 = -1$. Все элементарные функции, рассмотренные выше, работают с комплексными числами. В этом случае возвращаемое значение также будет комплексным. В табл. 3.6 приведены наиболее часто используемые функции для работы с комплексными числами.

Для примера присвоим переменной x значение $1 + i$ и выполним над этой переменной несколько простых операций, в том числе выделим действительную и мнимую части. Напоминаем, что одинарная кавычка ''' обозначает в Scilab оператор эрмитова сопряжения, который при воздействии на скаляр эквивалентен обычному комплексному сопряжению.

```
-->x = 1 + %i
x =
1. + i
--> isreal (x)
ans =
F
-->x'
ans =
1. - i
-->y = 1 - %i
```

```

y =
1. - i
-->real (y)
ans =
1.
-->imag (y)
ans =
- 1.

```

Наконец, проверим выполнение равенства $(1 + i)(1 - i) = 1 - i^2 = 2$:

```

-->x * y
ans =
2.

```

3.8 Целые числа

Таблица 3.7 — Функции Scilab для создания целочисленных переменных

int8	int16	int32	uint8	uint16	uint32
------	-------	-------	-------	--------	--------

В Scilab определены шесть типов целочисленных переменных, для создания которых используются функции из табл. 3.7. В этом разделе мы рассмотрим основные свойства целых чисел и диапазоны их значений, обратимся к преобразованию целых типов, а также уделим внимание проблемам, связанным с выходом значения за пределы диапазона, и вопросам переносимости программного кода, использующего целые числа.

3.8.1 Обзор целых чисел

Тип целочисленной переменной определяет количество бит, которое используется для представления значения переменной в памяти. В свою очередь, количество используемых бит задает диапазон допустимых значений переменной:

1) целое число со знаком, для представления которого используется n бит, может принимать значения из диапазона $[-2^{n-1}; 2^{n-1} - 1]$,

2) беззнаковое целое число, занимающее n бит в памяти, может принимать значения из диапазона $[0; 2^n - 1]$.

Таблица 3.8 — Диапазоны значений целых чисел в Scilab

$y = \text{int8}(x)$	8-битовое число со знаком $[-2^7; 2^7 - 1] = [-128; 127]$
$y = \text{uint8}(x)$	8-битовое число без знака $[0; 2^8 - 1] = [0; 255]$
$y = \text{int16}(x)$	16-битовое число со знаком $[-2^{15}; 2^{15} - 1] = [-32768; 32767]$
$y = \text{uint16}(x)$	16-битовое число без знака $[0; 2^{16} - 1] = [0; 65535]$
$y = \text{int32}(x)$	32-битовое число со знаком $[-2^{31}; 2^{31} - 1] = [-2147483648; 2147483647]$
$y = \text{uint32}(x)$	32-битовое число без знака $[0; 2^{32} - 1] = [0; 4294967295]$

Таблица 3.9 — Функции преобразования целочисленных типов в Scilab

Iconvert	преобразование к целочисленному представлению
inttype	определение типа целого числа

Например, 8-битовое целое число со знаком (`int8`) может принимать значение в пределах $[-128; 127]$. Таблица 3.8 отражает соответствие между типом целого числа и диапазоном его значений.

Следующий фрагмент позволяет убедиться, что значение 32-битового целого числа без знака заключено в диапазоне $[0; 2^{32} - 1]$, т. е. от 0 до 4294967295:

```
--> format (25)
-->n = 32
n =
32.
-->2 ^ n - 1
ans =
4294967295.
-->i = uint32 (0)
i =
0
-->j = i - 1
j =
4294967295
-->k = j + 1
k =
0
```


3.8.2 Преобразование целых типов

Scilab предоставляет две функции для работы с типами целочисленных переменных (табл. 3.9). Функция `inttype` позволяет узнать тип переменной, содержащей целое число, возвращая одно из значений, перечисленных в таблице 3.10. При выполнении арифметических операций над целыми числами типом результирующего значения становится наиболее широкий из типов операндов. В следующем фрагменте показан пример сложения 8-битового числа `i` (для которого `inttype` возвращает 1) и 16-битового значения `j` (значение `inttype` равно 2). Полученное значение `k` имеет тип 16-битового целого числа.

```
-->i = int8 (1)
i =
1
--> inttype (i)
ans =
1.
-->j = int16 (2)
j =
2
--> inttype (j)
ans =
2.
-->k = i + j
k =
3
--> inttype (k)
ans =
2.
```

Таблица 3.10 — Коды типов, возвращаемые функцией `inttype`

<code>inttype(x)</code>	Тип переменной
1	8-битовое число со знаком
2	16-битовое число со знаком
4	32-битовое число со знаком
11	8-битовое число без знака
12	16-битовое число без знака
14	32-битовое число без знака

3.8.3 Выход за пределы диапазона и проблемы переносимости

Поведение целых чисел при выходе за границы допустимого диапазона заслуживает особого внимания, поскольку часто различается в пакетах математических вычислений. В Scilab значения целых чисел изменяются циклически, поэтому при увеличении на 1 за максимальным будет следовать минимальное значение диапазона.

Приведенный фрагмент позволяет в этом убедиться:

```
--> uint8 (0 + (-4 : 4))
ans =
252 253 254 255 0 1 2 3 4
--> uint8 (2 ^ 8 + (-4 : 4))
ans =
252 253 254 255 0 1 2 3 4
-->int8 (2 ^ 7 + (-4 : 4))
ans =
124 125 126 127 -128 -127 -126 -125 -124
```

Циклическое поведение целых чисел в Scilab способствует большей гибкости в работе, поскольку позволяет избежать избыточных проверок if. Вместе с тем алгоритмы, допускающие выход за пределы диапазона, должны проверяться с особой тщательностью, в особенности при переносе алгоритма из одной вычислительной среды в другую.

3.9 Целые числа и числа с плавающей точкой

По умолчанию числовая переменная в Scilab имеет вещественный тип с плавающей точкой. Для представления значений этого типа в памяти компьютера используется 64 бита. Даже если при создании переменной ей присваивается целое значение, переменная будет иметь тип с плавающей точкой. Целые числа, хранимые таким образом, получили название "flint" (сокращение от "floating point integer"). На практике целые значения в диапазоне [-252; 252] можно без особых опасений хранить в вещественных

переменных, так как Scilab в этом случае гарантирует точность выполнения операций.

Например, в следующем фрагменте мы производим точное сложение двух больших целых чисел, лежащих в пределах диапазона $[-252; 252]$, и убеждаемся, что потери точности не происходит:

```
-->a = 2 ^ 40 - 12
a =
1099511627764.
-->b = 2 ^ 45 + 3
b =
35184372088835.
-->c = a + b
c =
36283883716599.
```

Если же в ходе вычислений значение выходит за пределы указанного диапазона, результаты могут быть весьма неожиданными. Например, как демонстрирует следующий фрагмент, при работе с величинами, превышающими 253, полученное значение оказывается четным независимо от значений операндов:

```
-->(2 ^ 53 + (1 : 10)) '
ans =
9007199254740992.
9007199254740994.
9007199254740996.
9007199254740996.
9007199254740996.
9007199254740996.
9007199254740998.
9007199254741000.
9007199254741000.
9007199254741000.
9007199254741002.
```

В следующем примере мы вычислим значение 252 с использованием арифметики с плавающей точкой и 16-битовых целых чисел. В первом случае переполнения не происходит, даже несмотря на то, что значение находится на границе диапазона 64-битовых чисел с плавающей точкой.

Напротив, во втором случае результат оказывается неверным, так как значение 252 не представимо в рамках 16 бит.

```
-->2 ^ 52
ans =
4503599627370496.
--> uint16 (2 ^ 52)
ans =
0
```

Вопросы доступа к элементам матрицы посредством вещественных индексов будут рассмотрены в разделе 4.15.

3.10 Переменная ans

Если в инструкции Scilab не указано, какой переменной следует присвоить результат вычисления, результат заносится в специальную переменную ans. Переменная ans, будучи инициализированной, может использоваться так же, как любая другая переменная в Scilab.

В следующем фрагменте мы вычисляем значение $\exp(3)$, не указывая, куда поместить результат, и затем убеждаемся, что значение сохранено в переменной

```
ans:
-->exp (3)
ans =
20.08553692318766792368
-->t = log ( ans )
t =
3.
```

Переменная ans, как правило, используется только в интерактивном режиме для вычисления некоторого численного результата без создания новой переменной. Недостатком такого способа является возможность потери ранее полученных значений в результате перезаписи переменной ans.

При разработке скриптов полагаться на использование `ans` не следует — вместо этого необходимо применять обычные переменные.

3.11 Строки

Переменной в Scilab можно присвоить строковое значение, заключив его в двойные кавычки `""`. Конкатенация (слияние) строк осуществляется с помощью оператора `+`. В следующем примере мы объявляем две строки и склеиваем их при помощи оператора `+`:

```
-->x = " foo "
x =
foo
-->y = " bar "
y =
bar
-->x + y
ans =
foobar
```

Для обработки строк Scilab предоставляет широкий набор возможностей, включая функции для работы с регулярными выражениями, однако в данном документе эти функции рассматриваться не будут.

3.12 Динамическая типизация переменных

Тип переменной в Scilab может динамически меняться в зависимости от присвоенного ей значения. Таким образом, возможно, к примеру, создать переменную, содержащую вещественное число, а затем присвоить этой переменной строковое значение:

```
-->x = 1
x =
1.
-->x + 1
ans =
2.
-->x = " foo "
```

```
x =  
foo  
-->x + " bar "  
ans =  
foobar
```

Необходимо еще раз подчеркнуть, что Scilab является нетипизированным языком, поэтому нет необходимости указывать тип переменной до присвоения ей значения, и более того, тип может изменяться в течение времени жизни переменной.

4 МАТРИЦЫ

Матрицы играют ключевую роль в Scilab. В данном разделе будет рассмотрено создание матрицы, изменение ее свойств, доступ к элементам матрицы и высокоуровневые операции, работающие с группами элементов.

4.1 Обзор

Базовой структурой данных в Scilab является матрица. Всякая матрица характеризуется своим размером (т. е. числом строк и столбцов) и типом содержащихся в ней значений. Элементами матрицы могут являться вещественные, комплексные или целые числа, логические значения, строки или полиномы. Если две матрицы имеют совпадающее число строк и столбцов, говорят, что они имеют одинаковый размер. Частным случаем матриц являются векторы, в которых число строк либо столбцов равно 1. Собственно скалярные величины в Scilab отсутствуют — скалярное значение представляется матрицей 1×1 . Поэтому в данной главе мы рассматриваем работу с матрицами, подразумевая, что аналогичные действия применимы и к векторам (т. е. матрицам $n \times 1$ или $1 \times n$), и скалярам (матрицам 1×1).

Scilab создавался в первую очередь для работы с матрицами вещественных значений и поэтому содержит большое число функций для работы с вещественнозначными матрицами. В числе задач проектирования Scilab также стояла оптимизация скорости выполнения таких операций. Для этого было разработано специальное внутреннее представление матриц, позволяющее интерпретатору эффективно манипулировать ими. Основные операции линейной алгебры, такие как сложение, вычитание, транспонирование и скалярное произведение, выполняются оптимизированными встроенными функциями. Эти операции обозначаются в Scilab символами "+", "-", "" и "*".

При использовании высокоуровневых операторов и функций практически отпадает необходимость в реализации циклов, которые, помимо прочего, выполняются существенно медленнее (от 10 до 100 раз), нежели встроенные функции. Данное свойство Scilab носит название векторизации. Для написания максимально эффективных алгоритмов в Scilab необходимо всегда пользоваться имеющимися высокоуровневыми возможностями, так чтобы каждой командой по возможности обрабатывалась целая матрица, а не один ее элемент.

Более сложные задачи линейной алгебры, такие как решение систем линейных уравнений $Ax = b$, различные виды разложений (например, гауссово разложение с перестановками $PA = LU$), поиск собственных значений и векторов, также выполняются встроенными оптимизированными функциями. Пользователю эти возможности доступны посредством операторов Scilab `/` и `\` , а также специальных функций (например, функции `sres`, вычисляющей собственные значения и векторы для заданной матрицы).

4.2 Создание вещественных матриц

Определение вещественной матрицы в Scilab выглядит следующим образом:

$$A = [a_{11} , a_{12} , \dots , a_{1n} ; a_{21} , a_{22} , \dots , a_{2n} ; \dots ; a_{n1} , a_{n2} , \dots , a_{nn}],$$

где квадратные скобки `[` и `]` обозначают начало и конец перечисления элементов матрицы, запятой `,` отделяются элементы матрицы, находящиеся в одной строке, точка с запятой `;` разделяет строки матрицы.

Пробелы при отделении значений запятыми не являются обязательными, но улучшают читаемость. Символы `...` обозначают пропущенные значения.

В следующем примере мы создаем матрицу 2_3, содержащую действительные значения:

```
-->A = [1 , 2 , 3 ; 4 , 5 , 6]
A =
1. 2. 3.
4. 5. 6.
```

Таблица 4.1 — Функции для создания матриц

Eye	единичная матрица
Linspace	вектор равноотстоящих значений
ones	матрица, все элементы которой равны 1
zeros	нулевая матрица
testmatrix	специальные типы матриц (Гильберта, Франка и пр.)
rand	генератор случайных чисел
grand	улучшенный генератор случайных чисел

Можно воспользоваться упрощенным синтаксисом, не требующим применения символов ”,” и ”;”. В этом случае матрица записывается построчно, а значения в пределах одной строки разделяются пробелом, как показано ниже:

```
A = [ a11 a12 ... a1n
      a21 a22 ... a2n
      ...
      an1 an2 ... ann ]
```

Это позволяет существенно облегчить ввод матриц и повысить его наглядность:

```
-->A = [1 2 3
-->4 5 6]
A =
1. 2. 3.
4. 5. 6.
```

Многострочная запись матриц полезна при выводе матрицы в файл, поскольку упрощает просмотр и проверку значений, а также обратное считывание матрицы в Scilab.

Несколько функций Scilab позволяют создавать матрицы заданного размера, принимая в качестве своих параметров значения числа строк и столбцов.

Среди таких функций (см. табл. 4.1) наиболее часто используются `eye`, `zeros` и `ones`:

```
-->A = ones (2, 3)
A =
1. 1. 1.
1. 1. 1.
```

4.3 Пустая матрица []

Пустую матрицу (размером 0_0) можно создать, используя пустые квадратные скобки, как показано в следующем примере:

```
-->A = []
A =
[]
```

Этот синтаксис также позволяет очистить содержимое ранее созданной матрицы, чтобы освободить занимаемую память:

```
-->A = ones (100 , 100);
-->A = []
A =
[]
```

4.4 Определение размера матрицы

Функции, приведенные в табл. 4.2, позволяют проверить и изменить размер матрицы. Вызов функции `size` для некоторой матрицы возвращает два параметра `nr` и `nc`, значения которых равны числу строк и столбцов в данной матрице:

```
-->A = ones (2, 3)
A =
1. 1. 1.
1. 1. 1.
-->[nr , nc] = size (A)
nc =
3.
nr =
2.
```

Функция `size` может быть полезна при определении собственных функций в тех случаях, когда обработка аргументов зависит от размерностей аргументов. В качестве примера можно привести функцию, вычисляющую норму, которая будет различным образом действовать при получении вектора и матрицы. Функция `size` также имеет альтернативный синтаксис:

```
nr = size ( A , sel )
```

Возвращаемое значение в этом случае определяется вторым параметром `sel` функции:

- при `sel=1` или `sel="r"` возвращается число строк,
- при `sel=2` или `sel="c"` возвращается число столбцов,
- при `sel="*"` возвращается общее число элементов в матрице, равное числу строк, умноженному на число столбцов.

В приведенном ниже фрагменте с использованием функции `size` подсчитывается общее число элементов в матрице:

```
-->A = ones (2, 3)
A =
1. 1. 1.
1. 1. 1.
-->size (A, "*")
ans =
6.
```

4.5 Работа с элементами матрицы

Scilab предоставляет несколько способов доступа к элементам матрицы A :

- 1) используя имя матрицы A , можно оперировать матрицей как целым,
- 2) для поэлементных манипуляций применяется запись $A(i,j)$,
- 3) для доступа к группе элементов, индексы которых лежат в некотором диапазоне, служит оператор ":" — он будет рассмотрен в следующем разделе.

Таблица 4.2 — Функции для проверки и изменения свойств матрицы

Size	определить размер матрицы
matrix	изменить размер матрицы
resize_matrix	создать новую матрицу заданного размера и скопировать в нее элементы из исходной матрицы

Для операций с матрицами используются имена содержащих эти матрицы переменных. Все элементарные алгебраические операции применимы к матрицам одинакового размера, как это показано ниже на примере вычисления суммы:

```
-->A = ones (2, 3)
A =
1. 1. 1.
1. 1. 1.
-->B = 2 * ones (2, 3)
B =
2. 2. 2.
2. 2. 2.
-->A + B
ans =
3. 3. 3.
3. 3. 3.
```

Получить доступ к отдельному элементу матрицы можно с помощью записи $A(i,j)$, при условии что i и j представляют собой допустимые значе-

ния индексов. В Scilab первый элемент имеет индекс 1, в отличие от многих языков программирования, где индексы элементов массива начинаются с 0. Допустим, к примеру, что A представляет собой матрицу $n_r \times n_c$, где n_r есть число строк, а n_c — число столбцов. Запись $A(i,j)$ имеет смысл только при $1 \leq i \leq n_r$ и $1 \leq j \leq n_c$. Если значение хотя бы одного из двух индексов не является допустимым, выдается сообщение об ошибке:

```
-->A = ones (2, 3)
A =
1. 1. 1.
1. 1. 1.
-->A(1, 1)
ans =
1.
-->A(12 , 1)
!-- error 21
Invalid index .
-->A(0, 1)
!-- error 21
Invalid index .
```

Прямой доступ к элементам матрицы с использованием синтаксиса $A(i,j)$ следует использовать только в том случае, если отсутствует соответствующая высокоуровневая функция Scilab. В большинстве случаев такая функция может быть найдена. Scilab предоставляет богатый набор простых и эффективных команд, основанных на возможностях векторизации. Одна из таких возможностей заключена в использовании оператора ":", который оказывается чрезвычайно полезным на практике.

4.6 Оператор ":"

Простейшая запись инструкции с оператором ":" выглядит следующим образом:

```
v = i : j
```

где i — это минимальное, а j — максимальное значение индекса, причем $i \leq j$.

Результатом является вектор $v = (i; i+1; \dots; j)$. Следующий пример демонстрирует генерацию вектора, элементы которого пробегают значения от 2 до 4:

```
-->v = 2 : 4
v =
2. 3. 4.
```

Расширенный вариант записи оператора ":" позволяет указать приращение индекса, называемое шагом:

```
v = i : s : j
```

где i и j имеют тот же смысл, что и ранее, а s представляет собой шаг.

Данная команда создает вектор $v = (i; i+s; i+2s; \dots; i+ns)$, где n — максимальное целое число, такое, что $i + ns \leq j$. Если $(j - i)$ делится нацело на s , то максимальное значение индекса совпадает с j , в остальных случаях $i+ns < j$. Величина шага s может быть как положительной, так и отрицательной.

В следующем примере создается вектор со значениями от 3 до 10 с шагом 2:

```
-->v = 3 : 2 : 10
v =
3. 5. 7. 9.
```

Последнее значение в данном случае равно 9, т. е. меньше $j=10$. Следующий фрагмент иллюстрирует результат действия оператора ":" при отрицательном значении шага. В первом случае мы создаем вектор значений, убывающих от 10 до 4. Во втором случае результатом выполнения оператора становится пустая матрица, поскольку начальное значение меньше конечного.

```
-->v = 10 : -2 : 3
v =
10. 8. 6. 4.
-->v = 3 : -2 : 10
v =
[]
```

Используя вектор значений индекса, можно обращаться к элементам матрицы в определенных диапазонах строк и столбцов:

$$A(i : j , k : l)$$

где i, j, k, l — начальные и конечные значения индексов.

Полный синтаксис $A(i:s;j:k:t:l)$ включает также задание шагов s и t . Предположим, что A является матрицей размера 4×5 . Пусть необходимо обратиться к элементам $a_{i;j}$, причем $i = 1; 2$, а $j = 3; 4$. В Scilab для этого потребуется всего одна инструкция, $A(1:2,3:4)$, как показано в следующем фрагменте:

```
-->A = testmatrix (" hilb ", 5)
A =
  25.    - 300.    1050.    -1400.    630.
- 300.    4800.   -18900.    26880.  -12600.
 1050.   -18900.    79380.  -117600.   56700.
-1400.    26880.  -117600.    179200.  -88200.
 630.   -12600.    56700.   -88200.   44100.
-->A(1 : 2, 3 : 4)
ans =
  1050. - 1400.
- 18900. 26880.
```

В некоторых случаях значения индексов должны вычисляться динамически (например, в цикле). Полученные в результате векторы могут использоваться для адресации элементов матрицы точно так же, как числовые литералы:

$$A(v_i , v_j),$$

Данная запись выделяет элементы матрицы A , индексы которых принадлежат векторам v_i и v_j . Следующий пример иллюстрирует эту возможность:

```
-->A = testmatrix (" hilb ", 5)
A =
    25.    - 300.    1050.    -1400.    630.
   - 300.    4800.   -18900.    26880.  -12600.
    1050.   -18900.    79380.  -117600.   56700.
   -1400.    26880.  -117600.    179200.  -88200.
    630.   -12600.    56700.   -88200.   44100.
-->vi = 1 : 2
vi =
1. 2.
-->vj = 3 : 4
vj =
3. 4.
-->A(vi , vj)
ans =
    1050. - 1400.
   - 18900.  26880.
-->vi = vi + 1
vi =
2. 3.
-->vj = vj + 1
vj =
4. 5.
-->A(vi , vj)
ans =
    26880. - 12600.
   - 117600.  56700.
```

Таблица 4.3 — Различные варианты использования оператора ":" для доступа к элементам матрицы

A	матрица целиком
A(:,j)	матрица целиком
A(i:j,k)	элементы матрицы в k-ом столбце с i-ой по j-ую строку
A(i,j:k)	элементы матрицы в i-ой строке с j-ого по k-ый столбец
A(i,:)	i-ая строка матрицы
A(:,j)	j-ый столбец матрицы

На основе оператора ":" может быть построено большое число комбинаций. Наиболее часто используемые из них представлены в табл. 4.3.

Например, следующий фрагмент демонстрирует использование оператора ":" для перестановки строк матрицы:

```
-->A = testmatrix (" hilb ", 3)
A =
   9.  -36.   30.
  -36.  192. -180.
   30. -180.  180.
-->A ([1 2], :) = A ([2 1], :)
A =
  -36.   192.  -180.
    9.   -36.    30.
   30.  -180.   180.
```

Обменять местами столбцы матрицы A можно при помощи инструкции $A(:, [31\ 2])$. В этом разделе нами были рассмотрены несколько важных с практической точки зрения случаев использования оператора ":". Оператор ":" повсеместно используется при написании алгоритмов, для которых скорость выполнения имеет решающее значение. Именно использованием оператора ":" достигается векторизация алгоритмов — одна из ключевых возможностей Scilab, которой уделяется особое внимание в данном руководстве.

4.7 Генерация единичной матрицы. Функция eye

Функция `eye` позволяет создать единичную матрицу необходимого размера. Название этой функции в английском произношении созвучно названию буквы I , используемой в англоязычной литературе для обозначения единичной матрицы, и было выбрано таким с целью избежать путаницы с традиционным обозначением индексной переменной i или мнимой единицы 1 .

Следующий фрагмент демонстрирует, как прибавить число 3 к диагональным элементам матрицы A :

```
-->A = ones (3, 3)
1
```

В Scilab мнимая единица является предопределенной константой и обозначается %i.

```
A =
1. 1. 1.
1. 1. 1.
1. 1. 1.
-->B = A + 3 * eye ( )
B =
4. 1. 1.
1. 4. 1.
1. 1. 4.
```

Далее мы создаем единичную матрицу B с использованием функции eye, причем размер создаваемой матрицы будет равен размеру матрицы A, переданной функции eye в качестве параметра. Сама матрица A при этом не изменяется:

```
-->A = ones (2, 2)
A =
1. 1.
1. 1.
-->B = eye (A)
B =
1. 0.
0. 1.
```

Наконец, используя синтаксис eye(m,n), можно явно указать количество строк m и столбцов n в генерируемой матрице.

4.8 Динамическое изменение размера матрицы

Матрицы могут динамически изменять свой размер в процессе выполнения программы, что позволяет им приспосабливаться к хранимым данным.

Для иллюстрации создадим матрицу 2 x 3:

```
-->A = [1 2 3; 4 5 6]
A =
1. 2. 3.
4. 5. 6.
```

Далее мы добавим в эту матрицу элемент с индексами (3; 1), равный 7. При этом в матрице будет создана третья строка, и первому элементу в ней будет присвоено указанное значение, а остальные элементы строки примут значение 0.

```
-->A(3, 1) = 7
A =
1. 2. 3.
4. 5. 6.
7. 0. 0.
```

Таким образом, размер матрицы может увеличиваться динамически. Далее мы покажем, что возможно и его динамическое уменьшение. Для этого, используя обозначение пустой матрицы [], удалим третий столбец матрицы A:

```
-->A(:, 3) = []
A =
1. 2.
4. 5.
7. 0.
```

Также можно полностью изменить размер матрицы посредством функции `matrix`. Эта функция трансформирует исходную матрицу в матрицу другого размера, копируя элементы столбец за столбцом. В следующем примере матрица A размером 3 x 2 преобразуется в вектор-строку с 6 элементами:

```
-->B = matrix (A, 1, 6)
B =
1. 4. 7. 2. 5. 0.
```

4.9 Оператор "\$"

Обычно при обращении к элементам матрицы отсчет индексов ведется от первой строки и первого столбца. Оператор "\$" позволяет адресовать элементы матрицы, отсчитывая индексы от последней строки или

столбца в зависимости от контекста. Различные варианты использования оператора "\$" приведены в табл. 4.4.

Таблица 4.4 — Использование оператора "\$" для доступа к элементам матрицы A размера nr _ nc

A(i,\$)	элемент на пересечении i-ой строки и последнего (nc-ого) столбца
A(\$,j)	элемент на пересечении последней (nr-ой) строки и j-ого столбца
A(\$-i,\$-j)	элемент на пересечении строки nr-i и столбца nc-j

Следующий фрагмент демонстрирует использование оператора "\$" для доступа к элементу $A(2,1) = A(nr-1,nc-2) = A(\$-1,\$-2)$ матрицы 3 x 3:

```
-->A= testmatrix (" hilb ", 3)
A =
    9.   - 36.   30.
 - 36.   192. - 180.
    30. - 180.   180.
-->A($ - 1, $ - 2)
ans =
 - 36.
```

Оператор "\$" также позволяет динамически увеличить размер матрицы. В следующем фрагменте в Гильбертову матрицу добавляется новая строка:

```
-->A($ + 1, :) = [1 2 3]
A =
    9.   - 36.   30.
 - 36.   192. - 180.
    30. - 180.   180.
    1.     2.     3.
```

Чаще всего оператор "\$" используется в составе выражения \$+1, которое позволяет добавлять в матрицу строки или столбцы. Такой способ оказывается удобным, поскольку избавляет от необходимости постоянно отслеживать текущее число строк и столбцов. Однако пользоваться им не-

обходимо аккуратно — лишь в тех случаях, когда размер матрицы неизвестен заранее. Поскольку при каждом добавлении строки или столбца происходит выделение памяти под матрицу большего размера и копирование всех элементов из исходной матрицы, злоупотребление этой возможностью может привести к существенному снижению производительности программы.

4.10 Арифметические операции

Все арифметические операторы, такие как "+", "-", "*", "/", работают с вещественными матрицами. Ниже мы рассмотрим действие каждого оператора в отдельности, дабы устранить любую причину для путаницы. Операции сложения "+" и вычитания "-" выполняются в соответствии с обычными правилами линейной алгебры.

В следующем примере показано сложение двух матриц размерности 2×2:

```
-->A = [1 2
-->3 4]
A =
1. 2.
3. 4.
-->B = [5 6
-->7 8]
B =
5. 6.
7. 8.
-->A + B
ans =
6. 8.
10. 12.
```

Сложение матриц возможно только в случае одинаковых размерностей обоих операндов. Приведенный ниже фрагмент демонстрирует попытку сложения матриц размерностей 2×2 и 2×3, приводящую к ошибке:

```

-->A = [1 2
-->3 4]
A =
1. 2.
3. 4.
-->B = [1 2 3
-->4 5 6]
B =
1. 2. 3.
4. 5. 6.
-->A + B
!-- error 8
Inconsistent addition .

```

Исключение составляет ситуация, когда одно из слагаемых является скаляром (т. е. матрицей 1×1). В этом случае значение скаляра будет прибавлено к каждому элементу второго операнда:

```

-->A = [1 2
-->3 4]
A =
1. 2.
3. 4.
-->A + 1
ans =
2. 3.
4. 5.

```

Таблица 4.5 — Элементарные матричные операции и их поэлементные варианты

+ сложение	.+ поэлементное сложение
- вычитание	.- поэлементное вычитание
* умножение	.* поэлементное умножение
/ деление справа	./ поэлементное деление справа
\ деление слева	.\ поэлементное деление слева
^ или ** возведение в степень	.^ поэлементное возведение в степень
' эрмитово сопряжение (комплексное сопряжение и транспонирование)	.' транспонирование без сопряжения

Элементарные операторы для работы с матрицами представлены в табл. 4.5. Оператор деления в Scilab реализован в двух вариантах: деление справа, обозначаемое символом `"/`, и деление слева, которому соответст-

вует символ "\". Результат деления справа $X = A/B = AB^{-1}$ представляет собой решение уравнения $XB = A$. Результат деления слева $X = A\backslash B = A^{-1}B$ является решением уравнения $AX = B$.

В случае, когда A не является квадратной матрицей, результат деления слева $A\backslash B$ дает решение соответствующей задачи о наименьших квадратах. В табл. 4.5 представлены также поэлементные операторы, работающие с отдельными элементами матриц. Эти операторы будут подробнее рассмотрены в следующем разделе.

4.11 Поэлементные операции

Точка "." перед соответствующим арифметическим оператором обозначает, что операция производится поэлементно. Например, при использовании обычного оператора умножения "*" элементы матрицы-произведения $C = A*B$ вычисляются по известной формуле:

$$C_{ij} = \sum_{k=1,n} A_{ik} B_{kj},$$

при этом каждый элемент представляет собой сумму произведений элементов строки первой матрицы и столбца второй матрицы.

В случае поэлементного умножения $C = A.*B$ элемент c_{ij} представляет собой простое произведение соответствующих элементов a_{ij} и b_{ij} .

Следующий фрагмент демонстрирует эти различия для двух конкретных матриц:

```
-->A = ones (2, 2)
A =
1. 1.
1. 1.
-->B = 2 * ones (2, 2)
B =
2. 2.
2. 2.
-->A * B
ans =
4. 4.
```

```

4. 4.
-->A .* B
ans =
2. 2.
2. 2.

```

Особый смысл придает точка "." оператору эрмитова сопряжения. Об этом пойдет речь в следующем разделе.

4.12 Эрмитово сопряжение и транспонирование

Может возникнуть некоторая путаница между операторами "' ' " и ".'", первый из которых сопрягает и транспонирует матрицу, а второй — только транспонирует ее. Для вещественнозначных матриц действие обоих операторов эквивалентно и сводится к транспонированию. Различия начинают проявляться, когда эти операторы применяются в отношении матриц, содержащих комплексные значения.

В случае комплекснозначных матриц эрмитово сопряжение матрицы Z посредством оператора "' ' " дает матрицу $A = Z'$, значения элементов которой определяются выражением $A_{jk} = X_{kj} - iY_{kj}$, где X и Y — матрицы действительных и вещественных частей Z соответственно, а i — мнимая единица. При транспонировании без сопряжения, выполняемом с использованием оператора ".'", значения элементов результирующей матрицы $A_{jk} = X_{kj} + iY_{kj}$ отличаются от полученных выше знаками мнимых частей.

В следующем примере операторы "' ' " и ".'" применяются к асимметричной комплекснозначной матрице, так что различия в их действии становятся очевидны:

```

-->A = [1 2; 3 4] + %i * [5 6; 7 8]
A =
1. + 5.i 2. + 6.i
3. + 7.i 4. + 8.i
-->A'
ans =
1. - 5.i 3. - 7.i
2. - 6.i 4. - 8.i

```



```
-->A.'
ans =
1. + 5.i 3. + 7.i
2. + 6.i 4. + 8.i
```

Далее демонстрируется применение обоих операторов в случае вещественнозначной матрицы — результаты одинаковы:

```
-->B = [1 2; 3 4]
B =
1. 2.
3. 4.
-->B'
ans =
1. 3.
2. 4.
-->B.'
ans =
1. 3.
2. 4.
```

Поскольку указанные различия часто становятся источником ошибок, необходимо четко усвоить назначение каждого из операторов и применять оператор ”.’” в случае, когда требуется только транспонирование (будь то вещественная или комплексная матрица), и оператор ”.’” — когда необходимо выполнить транспонирование и сопряжение.

4.13 Умножение векторов

Допустим, $u \in \mathbb{R}^n$ является вектором-столбцом, а $v^T \in \mathbb{R}^n$ — вектором-строкой. Тогда элементы матрицы $A = uv^T$ представляют собой произведения $A_{ij} = u_i v_j$. Следующий фрагмент демонстрирует вычисление матрицы A :

```
-->u = [1
-->2
-->3]
u =
1.
2.
3.
-->v = [4 5 6]
```

```

v =
4. 5. 6.
-->u * v
ans =
  4.   5.   6.
  8.  10.  12.
 12.  15.  18.

```

Обычно в курсах линейной алгебры рассматриваются только векторы-столбцы, обозначаемые наподобие $u \in \mathbb{R}^n$. В этом случае соответствующий вектор-строка будет записан как u^T . В Scilab переменная может содержать непосредственно вектор-строку, а значит, при умножении транспонировать этот вектор не понадобится.

Также источником ошибок может служить и обратное предположение, что в переменной хранится вектор-строка, в то время как на самом деле переменная содержит вектор-столбец. Поэтому всякий алгоритм, работающий только с определенным типом матриц, должен проверять размерности входных аргументов и генерировать сообщение об ошибке в случае несоответствия.

4.14 Сравнение вещественных матриц

Сравнение двух матриц возможно при условии одинаковых размеров. При выполнении этого условия операторы, представленные в табл. 3.5, могут применяться и к матричным операндам. В результате сравнения двух матриц образуется матрица логических значений, где каждый элемент представляет результат сравнения соответствующих элементов исходных матриц. Логические операторы "&", "|" и др. также применимы к матрицам. Кроме того, матрицы логических значений могут выступать в качестве параметров функций `and` и `or`, смысл которых раскрывает табл. 4.6.

Таблица 4.6 — Функции построчного и постолбцового сравнения элементов матрицы

<code>and(A,"r")</code>	постолбцовое "И"
<code>and(A,"c")</code>	построчное "И"
<code>or(A,"r")</code>	постолбцовое "ИЛИ"
<code>or(A,"c")</code>	построчное "ИЛИ"

В следующем примере мы определяем матрицу A и сравниваем ее с числом 3 (при этом каждый элемент матрицы сравнивается с этим значением). Затем мы создаем вторую матрицу B и сравниваем ее с первой. Наконец, используя функцию `or`, мы выполняем построчное сравнение, в результате которого получаем вектор логических значений, указывающий, какие из столбцов матрицы A содержат элементы, превосходящие соответствующие элементы матрицы B.

```
-->A = [1 2 7
-->6 9 8]
A =
1. 2. 7.
6. 9. 8.
-->A > 3
ans =
F F T
T T T
-->B = [4 5 6
-->7 8 9]
B =
4. 5. 6.
7. 8. 9.
-->A > B
ans =
F F T
F T F
-->or(A > B, "r")
ans =
F T T
```

4.15 Числа с плавающей точкой в качестве индексов

Предположим, что матрица A имеет размер 2×2 . Для того чтобы обратиться к элементам матрицы, могут применяться как жестко заданные в тексте программы индексы, так и значения, вычисляемые динамически в ходе выполнения, в том числе нецелые, как показано далее:

```
-->A = testmatrix (" hilb ", 2)
A =
    4. - 6.
 - 6. 12.
-->A( 2 , [1.0 1.1 1.5 1.9] )
ans =
 - 6. - 6. - 6. - 6.
```

Этот пример демонстрирует, что все значения 1.0, 1.1, 1.5 и 1.9 округляются до 1, как при использовании функции `int`, усекающей дробную часть числа. Например, результатом каждого из следующих вызовов: `int(1.0)`, `int(1.1)`, `int(1.5)` и `int(1.9)` — будет значение 1, а выражения `int(-1.0)`, `int(-1.1)`, `int(-1.5)` и `int(-1.9)` возвращают число -1.

Заметим, что округление происходит в соответствии с особенностями работы функции `int`, а не `floor`, которая в отличие от `int` отрицательные значения округляет в меньшую сторону. В самом деле, предположим, что A представляет матрицу 4×4 , например созданную в результате вызова:

```
A = testmatrix (" hilb ", 4).
```

Функция `triu(A,k)` возвращает верхнюю треугольную часть переданной ей матрицы A , лежащую выше k -ой диагонали. Команды `triu(A, -1)`, `triu(A, int(-1.5))` и `triu(A, -1.5)` возвращают один и тот же результат, в то время как инструкция `triu(A, floor(-1.5))` эквивалентна `triu(A, -2)`.

Такое поведение может показаться странным, но способствует единообразию языка Scilab. Действительно, при обращении к элементам матрицы с использованием переменных в качестве индексов, типом перемен-

ной чаще всего будет являться число с плавающей точкой. В этом случае преобразование значения происходит посредством функции `int`. Сказанное иллюстрирует следующий фрагмент:

```
-->i = 2
i =
2.
-->j = 1
j =
1.
-->A(i, j)
ans =
- 6.
```

Обратите внимание, что в этом примере переменные `i` и `j` имеют тип с плавающей точкой. Иногда использование чисел с плавающей точкой в качестве индексов может приводить к неожиданным результатам, например:

```
-->ones (1, 1)
ans =
1.
-->ones (1, (1 - 0.9) * 10)
ans =
[]
```

При отсутствии ошибок округления, обусловленных конечной разрядностью представления чисел в памяти компьютера, значение выражения $(1 - .9) * 10$ равнялось бы 1, и в приведенном примере мы получили бы матрицу 1 x 1. Однако в действительности команда `ones(1,(1-0.9)*10)` возвращает пустую матрицу, так как в результате округления числа $(1-0.9)*10$ с использованием функции `int` получаем 0:

```
-->int ((1 - 0.9) * 10)
ans =
0.
```

Причиной тому являются особенности двоичного представления дробных чисел. В результате двоичное представление числа $1 - 0.9$ несколько меньше 0.1, а будучи умноженным на 10, дает число, которое меньше 1, в чем позволяет убедиться следующий фрагмент:

```
--> format (25)
-->1 - 0.9
ans =
0.09999999999999999777955
-->(1 - 0.9) * 10
ans =
0.99999999999999997779554
```

На практике при вычислении индексов не следует применять числа, которые не имеют точного представления в форме с плавающей точкой. Точные результаты дают обычные арифметические операции с целыми числами в пределах диапазона $[-252; 252]$.

4.16 Элементарные функции

В данном разделе рассматриваются тригонометрические функции, принимающие входные значения в градусах, логарифмические функции, а также специальные матричные операции. Тригонометрические функции, такие как \sin и \cos , принимают на вход значения в радианах. Их аналоги, предназначенные для работы с аргументами, выраженными в градусах, отличаются наличием буквы "d" в конце имени (например, sind , cosd).

Математически пары таких функций связаны соотношениями вида $\text{sind } x = \sin (x\pi/180)$. Однако ввиду особенностей компьютерной арифметики результаты выполнения функций для математически эквивалентных аргументов могут различаться.

Например, следующий фрагмент демонстрирует вычисление значений $\sin \pi$ и $\sin 180$:

```
-->sin ( %pi )
ans =
1.225D -16
-->sind (180)
ans =
0.
```

Тот факт, что вычисленное значение $\sin \pi$ не равно в точности 0, связан с ограниченной точностью представления действительных чисел в памяти компьютера. Действительно, количество разрядов в двоичном представлении числа π в памяти компьютера ограничено. Следовательно, и результат вычисления функции оказывается приближенным. В то же время число 180 представляется точно, поскольку является целым. Поэтому значение $\text{sin}(180)$ вычисляется точно.

Функция \log вычисляет натуральный логарифм числа — функцию, обратную экспоненте $\exp = e^x$, где e называется основанием натурального логарифма или константой Эйлера. Для вычисления логарифма по основаниям 10 и 2 можно использовать функции \log_{10} и \log_2 соответственно. В следующем примере мы вычислим значения \log , \log_{10} и \log_2 для набора заданных значений x :

```
-->x = [ exp (1) exp (2) 1 10 2^1 2^10]
x =
2.7182818 7.3890561 1. 10. 2. 1024.
-->[x' log (x ') log10 (x ') log2 (x ')]
ans =
    2.7182818    1.          0.4342945    1.442695
    7.3890561    2.          0.8685890    2.8853901
    1.           0.          0.           0.
    10.          2.3025851    1.           3.3219281
    2.           0.6931472    0.30103 1.
    1024.        6.9314718    3.0103 10.
```

В первом столбце отображены исходные значения x , второй столбец содержит значения $\log(x)$, третий и четвертый, соответственно, значения $\log_{10}(x)$ и $\log_2(x)$. Большинство функций являются поэлементными, т. е., получив на вход матрицу, воздействуют на каждый ее элемент независимо. Вместе с тем некоторые функции имеют специальное значение в линейной алгебре. Так, например, матричная экспонента определена как

$$e^X = \sum_{k=0, \infty} \frac{1}{k!} X^k.$$

Для расчета матричной экспоненты в Scilab применяется функция `expm`. Вполне очевидно, что применение обычной функции `exp` дает совершенно иной результат. Имена функций, имеющих специальное значение применительно к матрицам, в Scilab оканчиваются буквой "m", например `expm`, `sinm` и др. Следующий фрагмент демонстрирует воздействие функций `sin` и `sinm` на матрицу `A` размером 2×2 , содержащую кратные $\pi/2$ значения:

```
-->A = [ %pi /2 %pi ; 2* %pi 3* %pi /2]
A =
1.5707963 3.1415927
6.2831853 4.712389
-->sin (A)
ans =
1. 1.225D -16
- 2.449D -16 - 1.
-->sinm (A)
ans =
- 0.3333333 0.6666667
1.3333333 0.3333333
```

Таблица 4.7 — Некоторые функции линейной алгебры

<code>chol</code>	разложение Холесского	<code>Lu</code>	LU-разложение с выбором опорного элемента
<code>companion</code>	сопровождающая матрица	<code>qr</code>	QR-разложение
<code>cond</code>	число обусловленности	<code>rcond</code>	обратное число обусловленности
<code>det</code>	определитель матрицы	<code>spec</code>	собственные значения и векторы
<code>inv</code>	обратная матрица	<code>svd</code>	разложение по сингулярным числам матрицы
<code>linsolve</code>	решение систем линейных уравнений	<code>testmatrix</code>	генерация специальных матриц (Гильберта, Франка и др.)
<code>lsq</code>	метод наименьших квадратов	<code>trace</code>	след матрицы

5 ОПЕРАТОРЫ ВЕТВЛЕНИЯ И ЦИКЛА

В этом разделе рассматриваются конструкции структурного программирования, доступные в Scilab, такие как условный оператор `if`, оператор выбора `select`, циклы `for` и `while`, а также инструкции `break` и `continue`, предназначенные для управления выполнением цикла.

5.1 Оператор `if`

Оператор `if` позволяет выполнить некоторый блок инструкций в случае истинности заданного условия. В качестве условия может выступать переменная логического типа или любое выражение, результатом вычисления которого является логическое значение. Блок выполняемых инструкций завершается ключевым словом `end`. В следующем примере мы отображаем строку "Привет!", если условие `%t` выполняется. Выражение `%t`, как мы знаем, представляет собой константу, обозначающую истину, поэтому данное условие всегда истинно и строка будет отображена в любом случае.

```
if ( %t ) then
disp ("Привет!")
end
```

В результате выполнения этого примера в консоль будет выведено:

```
Привет!
```

Если условие ложно, выполняется ветвь `else`, как показано в следующем фрагменте:

```
if ( %f ) then
disp ("Привет!")
else
disp ("Пока!")
end
```

В данном случае в консоли появится строка:

Пока!

В качестве условия может использоваться любое выражение, результатом вычисления которого является логическое значение, например содержащие операторы сравнения "==" , ">" и т. д. или функции, возвращающие логическое значение. В следующем примере мы используем оператор "==" для проверки условия и выводим сообщение "Привет!", если условие истинно, и "Пока!", если оно ложно:

```
i = 2
if ( i == 2 ) then
disp ("Привет!")
else
disp ("Пока!")
end
```

Подчеркнем, что для проверки на равенство используется оператор "==" , а не "=" . Использование последнего приводит к предупреждению, как показано в следующем примере:

```
-->i = 2
i =
2.
-->if ( i = 2 ) then
Warning : obsolete use of '=' instead of '==' '.
!
--> disp ("Привет!")
Hello !
-->else
--> disp ("Пока!")
-->end
```

При наличии нескольких условий, которые должны быть проверены последовательно, полезной оказывается конструкция elseif. Следующий фрагмент демонстрирует, каким образом конструкция elseif может применяться для обработки различных значений переменной i:

```

i = 2
if ( i == 1 ) then
disp ("Привет!")
elseif ( i == 2 ) then
disp ("Пока!")
elseif ( i == 3 ) then
disp ("До свидания!")
else
disp ("Бай!")
end

```

Количество блоков `elseif` не ограничено, что позволяет создавать ветвление произвольной сложности. Однако большое число следующих друг за другом блоков `elseif` часто говорит о необходимости использования оператора `select`, речь о котором пойдет в следующем разделе.

5.2 Оператор `select`

Оператор `select` предназначен для сокращенной записи нескольких последовательных проверок переменной на равенство одному из ряда значений, которые в противном случае необходимо было бы оформить как блоки `elseif`. В зависимости от значения переменной оператор `select` выполняет один из блоков `case`. Количество таких блоков не ограничено.

Ниже показано, как отобразить одну из нескольких возможных строк в соответствии со значением переменной `i`:

```

i = 2
select i
case 1
disp ("Один")
case 2
disp ("Два")
case 3
disp ("Три")
else
disp ("Иное значение")
end

```

В результате, как и следовало ожидать, в консоли будет отображена строка "Два".

Блок `else` выполняется в том случае, если значение переменной не соответствует ни одному из перечисленных вариантов. Наличие блока `else` не является обязательным, но считается хорошим тоном. Действительно, даже если программист полагает, что соответствующее этому блоку событие в нормальных условиях никогда не может произойти, ошибка в логике выполнения программы может привести к непредсказуемым последствиям. При отсутствии проверки скрипт продолжит выполняться и в наихудшем случае завершится без сообщений об ошибках, вернув неверный результат. Отладка такого скрипта представляет крайне сложную задачу, поскольку неясно, какая инструкция повлекла нарушение работы. Блок `else` призван воспрепятствовать распространению ошибки и может помочь в том, чтобы точнее определить место ее возникновения.

Таким образом, блок `else` должен присутствовать в большинстве конструкций `select`. Для обработки непредвиденных ситуаций в составе блока `else` часто применяется функция `error`. Функция `error` отображает сообщение об ошибке, содержащее указанный в качестве параметра текст. При этом выполнение алгоритма прерывается, интерпретатор Scilab покидает все вызванные функции и возвращает управление консоли.

Модифицируем предыдущий пример. Теперь в случае отрицательного значения переменной `i` будем выводить сообщение об ошибке:

```
i = -5;
select i
case 1
disp ("Один")
case 2
disp ("Два")
case 3
disp ("Три")
else
error ( "Непредусмотренное значение параметра i" )
end
```

В результате будет отображено:

```
-->i = -5;
--> select i
-->case 1
--> disp ("Один")
-->case 2
--> disp ("Два")
-->case 3
--> disp ("Три")
-->else
--> error ( "Непредусмотренное значение параметра i" )
Непредусмотренное значение параметра
i
```

Чаще всего, когда конструкция `select` не содержит блока `else`, следует задаться вопросом, является ли такая ситуация результатом тщательного анализа либо, наоборот, банального упущения. Предположение о том, что блок `else` никогда не будет выполнен, в большинстве случаев не оправдывается.

5.3 Оператор `for`

Оператор `for` применяется для повторения некоторого действия заданное число раз. Чаще всего используется числовой счетчик, пробегающий ряд значений. В конце этого раздела мы увидим, что цикл `for` является гораздо более универсальным и позволяет проводить итерации по элементам матрицы произвольного типа.

Следующий фрагмент печатает значения `i` от 1 до 5:

```
for i = 1 : 5
disp (i)
end
```

В результате в консоль будет выведено:

```
1.
2.
3.
4.
5.
```

Использование оператора ":" для генерации вектора [1 2 3 4 5] было рассмотрено в разделе 4.6. Запись 1:5 в приведенном примере создает матрицу, содержащую пять вещественных значений:

```
-->i = 1 : 5
i =
1. 2. 3. 4. 5.
```

Стоит еще раз подчеркнуть, что в данном случае матрица 1:5 содержит вещественные значения. Поэтому переменная-счетчик *i* также является вещественной. К этому вопросу мы вернемся позже в данном разделе, рассматривая общую форму записи цикла `for`. Используя более общую форму записи оператора ":", можно вывести только нечетные числа в интервале от 1 до 5. Для этого, очевидно, в качестве шага необходимо задать значение 2:

```
for i = 1 : 2 : 5
disp (i)
end
```

В результате выполнения этого алгоритма в консоли будут отображены строки:

```
1.
3.
5.
```

Оператор ":" можно также использовать для перебора значений счетчика в порядке убывания. Следующий фрагмент отображает числа от 5 до 1 в порядке убывания:

```
for i = 5 : -1 : 1
disp (i)
end
```

В консоль, как и следовало ожидать, будет выведено:

- 5.
- 4.
- 3.
- 2.
- 1.

Действительно, инструкция `5:-1:1` возвращает вектор значений от 5 до 1 в порядке убывания:

```
-->i = 5 : -1 : 1
i =
5. 4. 3. 2. 1.
```

Как уже говорилось, цикл `for` является гораздо более универсальным, позволяя перебирать значения различных типов, в том числе матриц и списков. В качестве элементов матриц могут выступать вещественные и целые числа, строки и полиномы.

В следующем примере цикл `for` используется для перебора элементов вектора-строки, содержащего вещественные значения (1.5 ; e ; π):

```
v = [1.5 exp (1) %pi ];
for x = v
disp (x)
end
```

В результате в консоли будет отображено:

```
1.5
2.7182818
3.1415927
```

Обращаем внимание читателей на нежелательность использования циклов в Scilab. Всякий раз, когда возникает такое намерение, следует выяснить, не существует ли подходящей векторизованной функции, выполняющей требуемые действия. Различие в скорости выполнения пользовательских циклов и встроенных функций может достигать 10–100 раз, поэтому при наличии такой функции предпочтение должно быть отдано ей. Цикл `for` следует использовать лишь в тех редких случаях, когда подходящей векторизованной функции найти не удастся.

5.4 Оператор while

Оператор `while` предназначен для повторения некоторого блока инструкций до тех пор, пока условие цикла остается истинным. Проверка условия выполняется перед каждой (в том числе первой) итерацией. В определенный момент условие повторения обращается в ложь и цикл завершается. Заметим, что для корректного завершения цикла необходимо, чтобы в теле цикла каким-либо образом изменялись переменные, входящие в условие продолжения, так, чтобы в какой-то момент значение этого выражения изменилось бы с истинного на ложное 1.

Следующий фрагмент демонстрирует применение цикла `while` для подсчета суммы чисел от 1 до 10:

```
s = 0
i = 1
while ( i <= 10 )
s = s + i
i = i + 1
end
```

Значения переменных по завершении выполнения этого фрагмента равны:

```
s =
55.
i =
11.
```

Заметим, что приведенный пример служит исключительно для иллюстрации работы цикла `while`. На практике, если потребуется вычислить сумму чисел от 1 до 10, следует воспользоваться функцией `sum`:

```
-->sum (1 : 10)
ans =
55.
```

Все сказанное в предыдущем разделе о недостатках цикла `for` в равной степени относится и к `while`. Поэтому векторизованные вычисления

также предпочтительны по сравнению с написанием собственных алгоритмов, использующих `while`.

5.5 Инструкции `break` и `continue`

Инструкция `break` позволяет прервать выполнение цикла. Обычно она применяется для выхода из цикла при достижении определенного условия, делающего его продолжение бессмысленным.

Следующий фрагмент демонстрирует использование инструкции `break` для вычисления суммы чисел от 1 до 10. При достижении переменной `i` значения, превышающего 10, цикл завершается:

```
s = 0
i = 1
while ( %t )
if ( i > 10 ) then
break
end
s = s + i
i = i + 1
end
```

После выполнения данного алгоритма значения переменных `s` и `i` равны:

```
s =
55.
i =
11.
```

Исключение составляют ситуации с использованием инструкции `break`, речь о которой пойдет в разделе 5.5.

Инструкция `continue` позволяет немедленно перейти к выполнению следующей итерации, пропустив команды, следующие после `continue` в теле цикла. Встретив команду `continue`, интерпретатор Scilab переходит к заголовку цикла, проверяет условие продолжения и, если оно истинно, делает следующую итерацию.

Следующий пример демонстрирует вычисление суммы $s = 1 + 3 + 5 + 7 + 9 = 25$. Используемая здесь функция `modulo(i,2)` возвращает 0 при условии, что i четно. В данном случае скрипт наращивает значение i и использует инструкцию `continue` для перехода к следующей итерации:

```
s = 0
i = 0
while ( i < 10 )
i = i + 1
if ( modulo ( i , 2 ) == 0 ) then
continue
end
s = s + i
end
```

Значения переменных s и i после выполнения данного скрипта равны

```
s =
25.
i =
11.
```

Тот же результат можно получить, используя единственную команду `sum` в сочетании с оператором `”:`”, что является примером векторизованных вычислений в Scilab:

```
s = sum (1 : 2 : 10);
```

Использование высокоуровневой функции (в данном случае `sum`) имеет ряд преимуществ перед эквивалентным ей с точки зрения результата циклом на основе `while`:

1. Высокоуровневая запись короче, а значит, проще для понимания человеком.

2. Для матриц значительной размерности высокоуровневые операции выполняются намного быстрее, чем алгоритмы на основе циклов.

Поэтому необходимо внимательно изучить доступные в Scilab функции перед тем, как писать собственный алгоритм с использованием `while`.

6 ФУНКЦИИ

Данный раздел посвящен функциям в Scilab. Здесь мы рассмотрим вопросы определения собственных функций и их загрузки в Scilab, научимся создавать библиотеки, представляющие собой наборы функций. Поскольку большая часть возможностей Scilab реализована в виде функций, мы увидим, как исследовать свойства той или иной функции. Также будет рассмотрена работа со входными и выходными аргументами. Наконец, мы обсудим возможности отладки функций с использованием инструкции `pause`.

6.1 Обзор

Выделение последовательности инструкций в отдельную функцию, пригодную для повторного использования, является одной из наиболее распространенных задач при работе со Scilab. Простейший синтаксис вызова функции выглядит следующим образом:

```
outvar = myfunction ( invar )
```

Значение каждого из трех элементов вызова функции приведено в списке:

- 1) `myfunction` представляет собой наименование вызываемой функции,
- 2) `invar` обозначает входные аргументы,
- 3) `outvar` соответствует выходным аргументам.

Значения переменных, указанных при вызове в качестве фактических параметров, функция изменить не может. Мы уже имели дело со многими функциями на протяжении данного руководства. Например, функция `sin` в составе команды `y = sin(x)` принимает входной аргумент `x` и помещает результат вычисления в переменную `y`. В соответствии с терминологией

Scilab входные аргументы называются правосторонними, а выходные — левосторонними.

Количество входных и выходных аргументов функции не ограничено. Синтаксис вызова функции с фиксированным числом аргументов таков:

```
[o1 , ... , on] = myfunction ( i1 , ... , in )
```

Список входных аргументов ограничивается круглыми скобками, а выходных — квадратными. Названия переменных в списках отделяются друг от друга запятыми “,”.

Следующий фрагмент демонстрирует выполнение LU-разложения матрицы Гильберта. Для начала матрица генерируется с использованием функции `testmatrix`, принимающей два входных аргумента: тип матрицы и ее порядок. Созданную матрицу мы передаем функции `lu`, которая возвращает две или три матрицы в зависимости от заданного пользователем количества выходных аргументов. Если аргументов три, в качестве последнего возвращается матрица перестановок `P`:

```
-->A = testmatrix (" hilb ", 2)
A =
  4. - 6.
 - 6. 12.
-->[L, U] = lu(A)
U =
 - 6. 12.
  0.  2.
L =
 - 0.6666667 1.
  1.          0.
-->[L, U, P] = lu(A)
P =
 0. 1.
 1. 0.
U =
 - 6. 12.
  0.  2.
L =
  1.          0.
 - 0.6666667 1.
```

Как мы могли убедиться, поведение функции `lu` зависит от числа выходных аргументов: во втором случае строки матрицы L меняются местами. Говоря точнее, при двух выходных аргументах выполняется разложение $A = LU$ (команда `A-L*U` позволяет проверить корректность результата), а при трех — разложение $PA = LU$ с матрицей перестановкой P (в чем можно убедиться, выполнив команду `P*A-L*U`). Таким образом, функция `lu` выбирает соответствующий алгоритм в зависимости от количества переданных ей параметров. Scilab предоставляет также возможность определения функций с переменным числом аргументов, однако этот вопрос выходит за рамки данного руководства.

Инструкции Scilab, предназначенные для работы с функциями, приведены в табл. 6.1. В последующих разделах наиболее часто используемые из них будут рассмотрены подробно.

Таблица 6.1 — Ключевые слова и инструкции Scilab, использующиеся при работе с функциями

<code>function</code>	открывает определение функции
<code>endfunction</code>	завершает определение функции
<code>argn</code>	количество входных и выходных аргументов в данном вызове функции
<code>varargin</code>	вектор, представляющий переменное число входных аргументов функции
<code>varargout</code>	вектор, представляющий переменное число выходных аргументов функции
<code>fun2string</code>	генерирует текстовое определение (исходный код) функции
<code>get_function_path</code>	возвращает путь к файлу исходного кода функции
<code>getd</code>	отображает полный список функций, определения которых хранятся в заданном каталоге файловой системы
<code>head_comments</code>	отображает комментарии к функции
<code>Listfunctions</code>	отображает свойства функций, которые вызывались ранее в данной сессии
<code>Macrovar</code>	возвращает списки входных и выходных параметров функции, а также используемых в теле функции внешних переменных, вызовов других функций и локальных переменных

6.2 Создание собственной функции

Для определения новой функции используются ключевые слова `function` и `endfunction`. В следующем примере мы создаем функцию `myfunction`, которая принимает единственный параметр `x`, умножает его значение на 2 и возвращает результат в качестве выходного параметра `y`:

```
function y = myfunction ( x )  
y = 2 * x  
endfunction
```

Строка `function y = myfunction (x)` представляет собой заголовок функции, в то время как тело функции содержит единственную инструкцию `y = 2*x`. В общем случае тело функции может включать произвольное число команд.

В Scilab существует, по меньшей мере, три способа определить такую функцию:

- 1) Во-первых, можно ввести тело функции непосредственно в консоли Scilab, инструкция за инструкцией. Встретив запись наподобие `function y = myfunction (x)`, интерпретатор переходит в режим ожидания тела функции. Завершается ввод командой `endfunction`, после чего Scilab возвращается в обычный режим.

- 2) Более удобным вариантом является определение функции в отдельном файле. Этот способ применяется в большинстве случаев. Для того чтобы загрузить заданную таким образом функцию, можно скопировать содержимое файла в консоль (удобно, если определение функции содержит всего несколько строк) либо воспользоваться командой Загрузить в Scilab (Load into Scilab) в меню Scilab.

- 3) Также для загрузки функции можно использовать команду `exec`. Предположим, что определение функции размещается в файле `C:\myscripts\examples-functions.sce`.

Для загрузки применяется команда `exec`, как показано в следующем фрагменте:

```
-->exec ("C:\myscripts \ examples - functions . sce ")
--> function y = myfunction ( x )
--> y = 2 * x
--> endfunction
```

Функция `exec` предназначена для исполнения инструкций, содержащихся в некотором файле, так, как если бы они вводились непосредственно в консоли Scilab. При этом в консоли отображается каждая строка алгоритма.

Если файл содержит большое число команд, отображение каждой из них может оказаться нежелательным. Чтобы этого избежать, после инструкции `exec` необходимо поставить ";" :

```
-->exec ("C:\myscripts \ examples - functions . sce " );
```

Того же результата можно добиться, выбрав пункт меню «выполнить файл в Scilab» (Execute file into Scilab). После того как функция создана, ее можно использовать подобно любой другой команде Scilab:

```
-->exec ("C:\myscripts \ examples - functions . sce ");
-->y = myfunction ( 3 )
y =
6.
```

Заметим, что присвоение выходному аргументу `y` значения (в данном случае $y=2*x$) является обязательным. Для того чтобы убедиться в этом, рассмотрим следующий пример, где значение присваивается переменной `z`, а не выходному параметру `y`:

```
function y = myfunction ( x )
z = 2 * x
endfunction
```

Попытаемся теперь вызвать эту функцию, передав ей значение 1:

```
--> myfunction ( 1 )  
!-- error 4  
Undefined variable : y  
at line 4 of function myfunction called by :  
myfunction ( 1 )
```

Интерпретатор Scilab сообщает нам об ошибке, поскольку переменная *y* не была инициализирована в теле функции.

При решении некоторой задачи часто возникает потребность в определении более чем одной функции. Например, при выполнении оптимизации с использованием функции *optim* нам, во-первых, требуется задать целевую функцию в соответствии с форматом, которого ожидает *optim*, а во-вторых — определить функцию-исполнитель, которая будет вызывать *optim* с требуемыми параметрами. В данном случае необходимы всего две функции, однако на практике для решения задачи может потребоваться и несколько десятков функций. В этом случае будет разумным объединить функции в библиотеку, как показано в следующем разделе.

6.3 Библиотеки функций

Библиотека представляет собой набор функций, написанных на языке Scilab и хранящихся в отдельных файлах. Если набор функций невелик и не содержит файлов справки или исходных текстов на компилируемых языках (таких как C/C++ или Fortran), объединение их в библиотеку является весьма удачным вариантом, в противном случае следует задуматься о создании модуля. Разработка собственного модуля не представляет трудностей, однако требует более детального знакомства с внутренним устройством пакета Scilab. Кроме того, модули также имеют в своей основе библиотеки, поэтому для создания первых требуется понимание работы последних. Тем не менее построение собственного модуля выходит за рамки настоящего руководства. Во многих практических ситуациях создание библиотеки является достаточным для организации набора функций.

В этом разделе рассматривается создание простой библиотеки функций Scilab, а также способы ее автоматической загрузки при запуске пакета. Предположим, что имеется несколько файлов .sci, содержащих определения функций на языке Scilab.

Таблица 6.2 — Инструкции Scilab для работы с библиотеками функций

genlib	создает библиотеку из функций, определения которых расположены в заданном каталоге
lib	загружает библиотеку

Последовательность действий в этом случае такова:

1. Создать бинарные версии функций, используя команду genlib. Функция genlib помимо прочего генерирует индексные файлы.

2. Загрузить библиотеку в Scilab, для чего используется функция lib.

Перед тем как приступить к рассмотрению примера, необходимо обозначить основные правила создания библиотек функций в Scilab:

1) Файлы, содержащие определения функций, должны иметь расширение .sci. Строго говоря, это требование не является обязательным, но помогает при поиске скриптов Scilab на жестком диске компьютера.

2) В одном файле .sci могут быть определены несколько функций Scilab, однако только первая из них будет доступна извне. Иными словами, только первая функция, определенная в файле, считается общедоступной, в то время как остальные неявно полагаются закрытыми (служебными) функциями.

3) Имя файла .sci должно совпадать с именем общедоступной функции в этом файле. Например, если имя функции myfun, то файл, содержащий ее, должен иметь название myfun.sci. Это требование является обязательным, в противном случае функция genlib не будет работать корректно. Инструкции Scilab, используемые при работе с библиотеками функций, представлены в табл. 6.2.

Теперь перейдем к примеру создания конкретной библиотеки. Предположим, что мы работаем на компьютере под управлением ОС Windows. Пусть в каталоге `samplelib` размещаются два файла:

```
_ C:/samplelib/function1.sci:
function y = function1 ( x )
y = 1 * function1_support ( x )
endfunction
function y = function1_support ( x )
y = 3 * x
endfunction
_ C:/samplelib/function2.sci:
function y = function2 ( x )
y = 2 * x
endfunction
```

Для получения бинарных версий этих функций используем инструкцию `genlib`. Первый аргумент `genlib` представляет название будущей библиотеки, а второй указывает каталог, где размещены файлы функций. Заметим, что в данном случае только функции `function1` и `function2` являются общедоступными, а функция `function1_support` может использоваться только внутри библиотеки, но не вне ее.

```
--> genlib ( " mylibrary ", "C:/ samplelib ")
--> mylibrary
mylibrary =
Functions files location : C:\ samplelib \.
function1 function2
```

Функция `genlib` генерирует и помещает в каталог `C:/samplelib` следующие файлы:

- `function1.bin`: бинарная версия файла `function1.sci`,
- `function2.bin`: бинарная версия файла `function2.sci`,
- `lib`: бинарная версия библиотеки,
- `names`: текстовый файл, содержащий имена всех функций в библиотеке.

Полученные файлы *.bin и файл lib являются кроссплатформенными в том смысле, что могут без изменений использоваться версиями Scilab для Windows, Linux и Mac OS.

Сразу же после вызова genlib две новые функции становятся доступны окружению Scilab и могут быть вызваны, как показано ниже:

```
--> function1 (3)
ans =
9.
--> function2 (3)
ans =
6.
```

Конечно, каждый раз генерировать библиотеку заново не требуется. Готовую библиотеку можно загрузить посредством команды lib, единственный аргумент которой указывает местоположение загружаемой библиотеки в файловой системе. Следующий фрагмент иллюстрирует загрузку созданной нами библиотеки:

```
--> mylibrary = lib ("C:\ samplelib \")
ans =
Functions files location : C:\ samplelib \.
function1 function2
```

При большом числе загружаемых библиотек удобно поместить вызовы lib в стартовый скрипт Scilab, который автоматически выполняется при запуске пакета. В этом случае все указанные библиотеки будут доступны сразу же после запуска Scilab. Файл стартового скрипта размещается в основном каталоге Scilab, путь к которому можно узнать, проверив значение переменной SCIHOME:

```
--> SCIHOME
SCIHOME =
C:\ Users \ username \ AppData \ Roaming \ Scilab \ scilab -
5.2.0
```

Стартовый скрипт носит имя `.scilab` и является обычным скриптом Scilab, в том числе может содержать комментарии. Для загрузки созданной ранее библиотеки добавим в файл `.scilab` следующие строки:

```
// Load my favorite library.
mylibrary = lib ("C:/ samplelib /")
```

В результате библиотека `mylibrary` будет загружаться всякий раз при запуске Scilab.

6.4 Управление выходными переменными

Каждая функция Scilab может иметь один или несколько входных и выходных аргументов. В простейшем случае число входных и выходных аргументов фиксировано, поэтому использовать такую функцию не составляет труда. Однако, как мы увидим далее, даже простейшие случаи допускают вариации.

Предположим, что функция `simplef` определена с двумя входными и двумя выходными аргументами:

```
function [y1 , y2] = simplef ( x1 , x2 )
y1 = 2 * x1
y2 = 3 * x2
endfunction
```

При вызове функции можно указать два, один либо ни одного выходного аргумента. Если не указано ни одного выходного аргумента, значение, которое должно было быть присвоено первому из них, помещается в переменную `ans`. Можно указать один выходной аргумент, который получит значение `y1`.

Наконец, можно предоставить функции два аргумента, как это предусмотрено ее определением. В следующем примере рассматриваются все три варианта вызова функции `simplef` с различным числом выходных переменных:

```

--> simplef ( 1 , 2 )
ans =
2.
-->y1 = simplef ( 1 , 2 )
y1 =
2.
-->[y1 ,y2] = simplef ( 1 , 2 )
y2 =
6.
y1 =
2.

```

Таким образом, даже простейшее определение функции допускает некоторую свободу относительно числа выходных аргументов. Более гибким способом задания переменного числа входных и выходных аргументов является использование ключевых слов `argn`, `varargin` и `varargout`. Эти возможности не рассматриваются в настоящем руководстве, однако для построения действительно гибких функций следует иметь их в виду.

6.5 Уровни стека вызовов

Как и в других языках программирования, в Scilab вызовы функций могут быть вложенными, т. е. функция `f` может вызывать функцию `g`, а та, в свою очередь, обращаться к функции `h` и т. д. Команды, введенные пользователем в консоли Scilab, соответствуют нулевому уровню стека вызовов. Инструкции, находящиеся в теле запущенной из консоли функции, составляют первый уровень. Каждый вложенный вызов увеличивает глубину стека на 1, так что текущая глубина равна длине цепочки вложенных вызовов. Функции, представленные в табл. 6.3, используются для получения информации о текущем состоянии стека.

Таблица 6.3 — Команды Scilab для работы со стеком вызовов

Whereami	отображает текущее дерево вызовов
Where	представляет текущее дерево вызовов в виде матрицы

Для иллюстрации мы определим три функции — `fmain`, `flevel1` и `flevel2`, вызывающие друг друга, и используем в `flevel2` инструкцию `whereami`, которая отображает текущее состояние стека вызовов:

```
function y = fmain ( x )
y = 2 * flevel1 ( x )
endfunction
function y = flevel1 ( x )
y = 2 * flevel2 ( x )
endfunction
function y = flevel2 ( x )
y = 2 * x
whereami ()
endfunction
```

При вызове функции `fmain` в консоль будет выведено:

```
--> fmain (1)
whereami called at line 3 of macro flevel2
flevel2 called at line 2 of macro flevel1
flevel1 called at line 2 of macro fmain
ans =
8.
```

Как можно видеть, отображаются три уровня стека вызовов, соответствующие вызванным функциям.

Значения уровней стека вызовов в приведенном примере таковы:

- уровень 0 — глобальный (команды, вводимые в консоли Scilab),
- уровень 1 — в теле функции `fmain`,
- уровень 2 — в теле функции `flevel1`,
- уровень 3 — в теле функции `flevel2`.

Эти уровни вызова отображаются в составе приглашения Scilab при интерактивной отладке функций с использованием инструкции `pause` и точек останова.

6.6 Инструкция return

Инструкция `return` позволяет завершить выполнение функции и вернуть управление вызывающему коду. Это бывает полезно, если в силу каких-то причин исполнение оставшейся части алгоритма потеряло смысл.

Представленная ниже функция вычисляет сумму целых чисел от `istart` до `iend`. При корректных значениях параметров для этого вызывается функция `sum`, однако если значение переменной `istart` отрицательно либо если условие `istart <= iend` не выполняется, значение переменной `y` устанавливается в 0 и функция завершает выполнение.

```
function y = mysum ( istart , iend )
if ( istart < 0 ) then
y = 0
return
end
if ( iend < istart ) then
y = 0
return
end
y = sum ( istart : iend )
endfunction
```

Следующий фрагмент позволяет убедиться, что функция `mysum` работает именно так, как задумано:

```
--> mysum ( 1 , 5 )
ans =
15.
--> mysum ( -1 , 5 )
ans =
0.
--> mysum ( 2 , 1 )
ans =
0.
```

Присутствие в теле одной функции многочисленных инструкций `return` считается плохим стилем программирования, поскольку серьезно усложняет анализ исходного кода такой функции. В этой связи рекоменду-

ется ограничиться использованием `return` лишь в тех ситуациях, где в противном случае потребовалось бы написание большого объема дополнительного кода. Как правило же, функция должна возвращать управление при достижении последней строки.

6.7 Отладка функций. Инструкция `pause`

В данном разделе мы рассмотрим основной метод отладки, позволяющий устранить большую часть простых ошибок. Этот метод основан на использовании инструкций `pause`, `resume` и `abort`, назначение которых раскрывает табл. 6.4.

Таблица 6.4 — Отладочные инструкции Scilab

<code>pause</code>	приостановить выполнение функции и ожидать команд пользователя
<code>resume</code>	продолжить выполнение функции
<code>abort</code>	завершить выполнение функции

Работа с пакетом Scilab обычно состоит в определении функций, реализующих алгоритмы для решения поставленной задачи. Однако работоспособность той или иной функции нередко оказывается под вопросом ввиду наличия синтаксических ошибок в ее коде.

Пусть требуется вычислить сумму целых чисел от `istart` до `iend`. Для этого определим функцию `mysum`, текст которой (приведен ниже) содержит ошибки. В частности, второй аргумент, "foo", передаваемый функции `sum`, избыточен.

```
function y = mysum ( istart , iend )
y = sum ( iend : istart , " foo " )
endfunction
```

Следующий фрагмент отражает результат выполнения функции `mysum`:


```
--> mysum ( 1 , 10 )
!-- error 44
Wrong argument 2.
at line 2 of function mysum called by :
mysum ( 1 , 10 )
```

Для того чтобы обнаружить проблему, поместим инструкцию `pause` в тело функции `mysum`:

```
function y = mysum ( istart , iend )
pause
y = sum ( iend : istart , " foo " )
endfunction
```

Снова обратимся к функции `mysum`, передав ей те же самые аргументы:

```
--> mysum ( 1 , 10 )
Type 'resume ' or 'abort ' to return to standard level prompt .
-1->
```

В данный момент мы находимся внутри тела функции `mysum`. Приглашение `"-1->"` указывает, что текущему положению соответствует уровень `-1` стека вызовов. Теперь мы можем проверить значения переменных `istart` и `iend`, введя их имена в консоли:

```
-1-> istart
istart =
1.
-1-> iend
iend =
10.
```

Чтобы увидеть, к какому результату приведет выполнение той или иной инструкции при текущих значениях переменных, скопируем эту инструкцию в консоль:

```
-1->y = sum ( iend : istart , " foo " )
y = sum ( iend : istart , " foo " )
!-- error 44
Wrong argument 2.
```

Теперь легко видеть, что источником проблемы является именно введенная строка. Наличие второго аргумента функции `sum` является в данном случае ошибкой, поэтому удалим его:

```
-1->y = sum ( iend : istart )
y =
0.
```

После этой модификации выполнение функции `mysum` больше не приводит к ошибкам, однако результат по-прежнему неверен. Можно заметить, что переменные `istart` и `iend` перепутаны местами. Проверим результат выполнения более корректного, на наш взгляд, вызова и убедимся, что он совпадает с ожидаемым:

```
-1->y = sum ( istart : iend )
y =
55.
```

Для выхода из режима отладки служит инструкция `abort`, которая прерывает всю цепочку вызовов и возвращает управление командной консоли Scilab.

```
-1-> abort
-->
```

Вид приглашения `-->` указывает, что теперь мы находимся на нулевом (глобальном) уровне стека вызовов.

Исправим определение функции в соответствии с выводами, сделанными в ходе отладки:

```
function y = mysum ( istart , iend )
pause
y = sum ( istart : iend )
endfunction
```

Вызовем функцию снова:

```
--> mysum ( 1 , 10 )
Type 'resume ' or 'abort ' to return to standard level prompt .
-1->
```

Поскольку теперь мы уверены в корректности кода, введем команду `resume`, которая указывает Scilab продолжить исполнение кода:

```
--> mysum ( 1 , 10 )  
-1-> resume  
ans =  
55.
```

Теперь, когда ошибки устранены, можно удалить инструкцию `pause` из тела функции:

```
function y = mysum ( istart , iend )  
y = sum ( istart : iend )  
endfunction
```

В данном разделе мы могли убедиться в эффективности интерактивной отладки функций с помощью команд `pause`, `resume` и `abort`. Приведенный пример, конечно, существенно упрощен и вряд ли требует отладки для обнаружения вполне очевидных ошибок. Тем не менее использование `pause` оказывается удобным и эффективным способом выявления неполадок и в гораздо более сложных ситуациях.

7 ПОСТРОЕНИЕ ГРАФИКОВ

Отображение графиков и других изображений является распространенной задачей при анализе данных и создании отчетов. Scilab предоставляет широкие возможности для создания и настройки различных типов графиков и диаграмм.

В данном разделе мы рассмотрим создание двухмерных и контурных графиков, затем отобразим на графике название и легенду и, наконец, увидим, как сохранить результаты отображения в файл для дальнейшего использования.

Таблица 7.1 — Функции Scilab для отображения графиков

plot	двухмерный график	barh	горизонтальная столбиковая диаграмма
surf	трехмерный график	hist3d	трехмерная гистограмма
contour	контурный график	polarplot	график в полярных координатах
pie	круговая диаграмма	matplot	цветной двухмерный график матрицы
histplot	Гистограмма	sgrayplot	сглаженный контурный график с использованием цвета
bar	столбиковая диаграмма	grayplot	несглаженный контурный график с использованием цвета

Таблица 7.2 — Вспомогательные функции, используемые при построении графиков

linspace	генерирует вектор из заданного числа равноотстоящих значений
feval	вычисляет значения функции в точках сетки
legend	задает легенду текущего графика
title	отображает название на текущем графике
xtitle	отображает название и подписи к осям на текущем графике

7.1 Обзор графических возможностей Scilab

Scilab предоставляет возможности для создания различных типов графиков, среди которых двухмерные, контурные и трехмерные графики, гистограммы, столбиковые и круговые диаграммы и др. Наиболее часто используемые функции для отображения графиков представлены в табл. 7.1.

Для того чтобы получить пример трехмерного графика, достаточно набрать в консоли Scilab команду `surf()`:

```
-->surf ( )
```

При создании графиков в данном разделе используются вспомогательные функции, приведенные в табл. 7.2.

7.2 Отображение двумерных графиков

В этом разделе мы увидим, как отобразить простой двумерный график функции, уделяя особое внимание возможностям векторизации, позволяющим создать матрицу исходных данных одной командой. Для начала определим функцию, график которой мы собираемся строить.

Наша функция `myquadratic` будет возводить свой аргумент x в квадрат, используя оператор `^`:

```
function f = myquadratic ( x )  
f = x ^ 2  
endfunction
```




Рис. 7.1 — График функции x^2

При помощи функции `linspace` создадим вектор из 50 равноотстоящих значений на отрезке [1; 10]:

```
xdata = linspace ( 1 , 10 , 50 );
```

Вектор `xdata` мы передаем функции `myquadratic`, которая рассчитывает значения в каждой точке:

```
ydata = myquadratic ( xdata );
```

Теперь в нашем распоряжении есть вектор-строка `ydata`, содержащий 50 элементов, который наряду с `xdata` мы используем в качестве параметра функции `plot` для отображения графика:

```
plot ( xdata , ydata )
```

Полученный график показан на рисунке 7.1. Отметим, что тот же график можно было построить без явного вычисления вектора значений `ydata`, передав функции `plot` в качестве второго параметра саму функцию `myfunction1`:

```
plot ( xdata , myquadratic )
```

Использование функции в качестве параметра позволяет существенно снизить затраты памяти при построении графика в случае большого числа точек.

7.3 Контурные графики

В этом разделе рассматриваются контурные графики функций двух переменных, для построения которых используется функция `contour`. Контурные графики находят активное применение при оптимизации функций, позволяя отобразить рельеф функции двух переменных так, что местонахождение оптимума становится очевидным.

Допустим, определена функция f от n переменных $f(x) = f(x_1; \dots; x_n)$ и $x \in \mathbb{R}^n$. Для заданного $\alpha \in \mathbb{R}$ уравнение:

$$f(x) = \alpha$$

определяет поверхность в $(n + 1)$ -мерном пространстве \mathbb{R}^{n+1} .

При $n = 2$ точки $z = f(x_1; x_2)$ образуют поверхность в трехмерном пространстве $(x_1; x_2; z) \in \mathbb{R}^3$, что позволяет отобразить контурный график целевой функции. При $n > 3$ столь удобного решения не существует — в этом случае можно выбрать две наиболее значимые переменные и построить график, варьируя только их.

Функция `contour`, позволяющая построить контурный график, имеет следующий синтаксис:

```
contour ( x , y , z , nz ),
```

где x и y — векторы-строки значений x и y , с числом элементов n_1 и n_2 соответственно; z — вещественнозначная матрица размером (n_1, n_2) , содержащая значения рассматриваемой функции, либо объект-функция Scilab, определяющая поверхность $z=f(x,y)$, nz — значения уровней либо их количество.

В следующем фрагменте мы используем простую форму функции `contour`, которой в качестве параметра передается функция `myquadratic`. Функция `myquadratic` принимает два аргумента x_1 и x_2 и возвращает значение $f(x_1; x_2) = x_1^2 + x_2^2$. Для генерации значений переменных, образующих сетку, используется функция `linspace`:

```
function f = myquadratic2arg ( x1 , x2 )
f = x1 ** 2 + x2 ** 2;
endfunction
xdata = linspace ( -1 , 1 , 100 );
ydata = linspace ( -1 , 1 , 100 );
contour ( xdata , ydata , myquadratic2arg , 10)
```

Полученный в результате график представлен на рис. 7.2. На практике функция, график которой необходимо отобразить, часто принимает единственный аргумент x , представляющий собой вектор-строку, в то время как функция `contour` требует наличия двух аргументов. Можно предложить следующие варианты решения данной проблемы:

- определить новую функцию, которая будет вызывать исходную,
- передать функции `contour` массив данных вместо объекта-функции.




Рис. 7.2 — Контурный график функции $f(x_1; x_2) = x_1^2 + x_2^2$

Оба этих подхода рассмотрены далее, так что читатель может выбрать наиболее подходящий вариант. Для начала обратимся ко второму способу, предполагающему генерацию массива значений функции. Пусть функция `myquadratic1arg` принимает на вход вектор из двух элементов. Для вычисления матрицы `zdata`, содержащей значения функции, выполняются два вложенных цикла. Для каждой комбинации $(x(i); y(j)) \in \mathbb{R}^2$ при $i = 1; 2; \dots; n_x$ и $j = 1; 2; \dots; n_y$, где n_x и n_y — это количество точек по осям x и y соответственно, в матрицу `zdata` заносится соответствующее значение. Наконец, для построения графика мы воспользуемся функцией

contour, передав ей список уровней (а не их число, как в предыдущем фрагменте). Это позволяет явно задать требуемые уровни вместо того, чтобы предоставлять Scilab их автоматическое вычисление.

```
function f = myquadraticlarg ( x )
f = x (1) ** 2 + x (2) ** 2;
endfunction
xdata = linspace ( -1 , 1 , 100 );
ydata = linspace ( -1 , 1 , 100 );
// Внимание! Применения двух вложенных циклов следует избегать.
for i = 1 : length ( xdata )
for j = 1 : length ( ydata )
x = [ xdata (i) ydata (j)]. ' ;
zdata ( i , j ) = myquadraticlarg ( x );
end
end
contour ( xdata , ydata , zdata , [0.1 0.3 0.5 0.7])
```

Полученный график показан на рис. 7.3. Рассмотренный фрагмент выполняет поставленную задачу, однако работает неэффективно из-за использования циклов. Для повышения скорости выполнения использования циклов следует предпочесть применение встроенных функций и векторизованных операций, рассмотренных ранее. В частности, для вычисления значений функции на сетке можно использовать функцию feval.




Рис. 7.3 — Контурный график функции $f(x_1; x_2) = x_1^2 + x_2^2$ с явным указанием уровней

Предположим, что модифицировать функцию `myquadratic1arg` невозможно, поэтому определим промежуточную функцию `myquadratic3`, принимающую два входных аргумента и вызывающую `myquadratic1arg`. Теперь, используя встроенную функцию `feval`, можем получить матрицу значений функции

```
zdata:
function f = myquadratic1arg ( x )
f = x (1) ** 2 + x (2) ** 2;
endfunction
function f = myquadratic3 ( x1 , x2 )
f = myquadratic1arg ( [x1 x2] )
endfunction
xdata = linspace ( -1 , 1 , 100 );
ydata = linspace ( -1 , 1 , 100 );
zdata = feval ( xdata , ydata , myquadratic3 );
contour ( xdata , ydata , zdata , [0.1 0.3 0.5 0.7])
```

Результатом является тот же контурный график, что и ранее (рис. 7.3). Наконец, построить график функции `myquadratic3` также можно, непосредственно передав эту функцию в качестве аргумента `contour`:

```
function f = myquadratic1arg ( x )
f = x (1) ** 2 + x (2) ** 2;
endfunction
function f = myquadratic3 ( x1 , x2 )
f = myquadratic1arg ( [x1 x2] )
endfunction
xdata = linspace ( -1 , 1 , 100 );
ydata = linspace ( -1 , 1 , 100 );
contour ( xdata , ydata , myquadratic3 , [0.1 0.3 0.5 0.7])
```

Полученный в результате график, конечно, будет в точности совпадать с предыдущими (рис. 7.2). Преимуществом же этого способа является экономия памяти, так как в данном случае нет нужды хранить матрицу значений `zdata`. Таким образом, мы вкратце рассмотрели построение возможности, которые Scilab предоставляет для отображения графиков. В следующем разделе мы обратимся к возможностям настройки таких элементов графика, как заголовок, названия осей и легенда.

7.4 Подписи на графиках

Для придания графику законченного вида необходимо отобразить название графика, подписи осей и легенду.

Вернемся к примеру построения графика функции x^2 , рассмотренному в разделе 7.2:

```
function f = myquadratic ( x )
f = x .^ 2
endfunction
xdata = linspace ( 1 , 10 , 50 );
ydata = myquadratic ( xdata );
plot ( xdata , ydata )
```

Результат выполнения данного фрагмента был показан на рис. 7.1. Графические возможности Scilab основаны на использовании графических дескрипторов. Графические дескрипторы предоставляют объектно-ориентированный доступ к свойствам графического объекта. Каждая графическая область состоит из примитивов, таких как линии, образующие кривые, оси, название графика, его легенда и т. п. Всякому графическому примитиву соответствует определенный набор свойств, например толщина, цвет линий и т. п. Доступ к этим свойствам осуществляется так же, как и к любым другим переменным Scilab.

Управление графическими дескрипторами, таким образом, является достаточно гибким и удобным механизмом работы с графическими объектами. Простейшее оформление графиков может выполняться посредством встроенных функций Scilab без явного обращения к дескрипторам — в данном руководстве мы ограничимся рассмотрением только этих базовых возможностей. Например, функция `title` используется для того, чтобы задать название графика:

```
title ( "Название графика" );
```

Для того чтобы отобразить на графике подписи осей, используем функцию `xtitle`:

```
xtitle ("Название графика", "Подпись оси X", "Подпись оси Y");
```

На рис. 7.4 представлен результат выполнения этой команды. Часто возникает необходимость отобразить совместно графики двух функций. Ниже мы определяем функции $f(x) = x^2$ и $f(x) = 2x^2$ и отображаем их на одном графике. Для того чтобы различать кривые, соответствующие каждой из функций, мы используем третий параметр функции `plot`. Значения "+" и "o" определяют способ отображения точек (в данном случае они будут отображаться символами "+" и "o") и соединяющих линий на графике (сплошная линия).




Рис. 7.4 — График функции x^2 с названием и подписями осей

```
function f = myquadratic ( x )
f = x ^ 2
endfunction
function f = myquadratic2 ( x )
f = 2 * x ^ 2
endfunction
xdata = linspace ( 1 , 10 , 50 );
```

```

ydata = myquadratic ( xdata );
plot ( xdata , ydata , "+-" )
ydata2 = myquadratic2 ( xdata );
plot ( xdata , ydata2 , "o-" )
xlabel ( "Название графика", "Подпись оси X", "Подпись оси Y");

```

Помимо этого необходимо отобразить легенду, указав, какая из кривых соответствует каждой из функций, для чего используется инструкция `legend`:

```
legend ( "x^2" , "2x^2" );
```

Рис. 7.5 демонстрирует получившийся в итоге график.




Рис. 7.5 — График двух функций $f_1(x) = x^2$ и $f_2(x) = 2x^2$.

Для соотнесения кривых и функций используются стили отображения точек и линий, а также легенда

Теперь мы знаем, как создать график и настроить параметры его отображения. Если график окажется достаточно интересным, его можно сохранить в файл для последующего использования. Экспорту графиков посвящен заключительный раздел данной главы.

7.5 Экспорт изображений

В данном разделе рассматриваются способы вывода изображения в графический файл с использованием оконного меню или встроенных функций Scilab.

Экспортировать выведенное на экран изображение можно, выбрав в меню пункт **Файл > Экспортировать...** (**File > Export to...**). Появляющееся после этого диалоговое окно предложит указать имя и тип создаваемого файла. Scilab позволяет экспортировать изображения в файлы многих распространенных векторных и растровых форматов. Для того чтобы сохранить изображение, не прибегая к помощи меню, используются функции, указанные в табл. 7.1.

Таблица 7.1 — Функции экспорта изображений

Векторные		Растровые	
xs2png	экспорт в формат PNG	xs2fig	экспорт в формат FIG
xs2pdf	экспорт в формат PDF	xs2gif	экспорт в формат GIF
xs2svg	экспорт в формат SVG	xs2jpg	экспорт в формат JPG
xs2eps	экспорт в формат Encapsulated Postscript	xs2bmp	экспорт в формат BMP
xs2ps	экспорт в формат Postscript	xs2ppm	экспорт в формат PPM
xs2emf	экспорт в формат EMF (только для Windows)		

Все функции, перечисленные в табл. 7.1, принимают два параметра, первый из которых, `window_number`, представляет собой номер графического окна (отображается в заголовке), а второй, `filename`, задает желаемое имя файла, куда будет записано изображение:

```
xs2png ( window_number , filename )
```

Следующая команда, к примеру, выведет содержимое окна с номером 0 в файл `foo.png`:

```
xs2png ( 0 , " foo . png " )
```

Для получения качественных документов предпочтительнее векторный формат изображений.

8 ЛАБОРАТОРНАЯ РАБОТА

8.1 Задание на лабораторную работу

1) Исследовать выбранное распределение плотности вероятности (РПВ) и функции распределения:

- а) построить распределение плотности вероятности;
- б) построить функцию распределения для трех значений изменения параметров попадающих в область изменения аргумента.

Пример ($X \in [-10, -9, \dots, 0, \dots, 9, 10]$, $a_1 = -3$, $a_2 = -9$, $a_3 = 7$).

Объяснить на графике поведение функции распределения.

2) Построить датчик случайных чисел (ДСЧ) для тех же параметров, принятых для построения плотности вероятности. Исследовать поведение ДСЧ для различного числа точек (10 и 100 и 1000) последовательности и изменения параметров функции.

3) Найти числа корреляции для трех типов (зависимость от изменения параметров) ДСЧ и объяснить поведение и зависимость кривых от выбранных параметров.

4) Построить матрицу корреляции и сравнить с пунктом (3), найти соответствие чисел корреляции и матрицы корреляции для каждого из случаев изменения параметров.

5) По выбранным трем ДСЧ построить гистограммы и сравнить их с истинной кривой плотности вероятности для одних и тех же заданных параметров. Указать, при каких значениях числа интервалов в гистограмме наблюдается максимальное соответствие плотностей вероятности.

б) Найти аддитивную смесь плотности вероятности и ДСЧ для одних и тех же параметров функции. Сгладить зашумленную кривую методом скользящего окна и методом медиан. Показать, при каких параметрах сглаживания наблюдается лучшее и точное сглаживание.

7) Вычитанием сглаженной функции из зашумленной найти случайную составляющую. Снова для случайной составляющей построить гистограмму, которую сравнить с РПВ для одних и тех же параметров.

8.2 Выбор варианта лабораторной работы

Выбор варианта лабораторной работы осуществляется по общим правилам с использованием следующей формулы:

$$V = (N \times K) \operatorname{div} 100,$$

где V — искомый номер варианта,

N — общее количество вариантов,

div — целочисленное деление,

при $V = 0$ выбирается максимальный вариант,

K — код варианта.

8.3 Варианты заданий

Вариант 1. ЭКСПОНЕНЦИАЛЬНОЕ РАСПРЕДЕЛЕНИЕ

$$p(x) = a1 \exp(-a1 * (x - a2))$$

$$a1 \in [0, \infty], \quad a2 \in [-\infty, \infty], \quad x \in [a2, \infty]$$

$$F(x) = \int_{-\infty}^{\infty} p(x) dx = 1 - \exp(-a1 * (x - a2))$$

$$R = a2 - \frac{1}{a1} \ln U$$

Вариант 2. ПОЛУНОРМАЛЬНОЕ РАСПРЕДЕЛЕНИЕ

$$p(x) = \frac{2}{a1\sqrt{2\pi}} \exp(-x^2 / (2a1^2))$$

$$a1 \in [0, \infty], \quad x \in [0, \infty]$$

$$F(x) = \int_{-\infty}^{\infty} p(x) dx = \frac{2}{\sqrt{2\pi}} \int_0^t \exp(-t^2 / 2) dt$$

$$R = a1 |Rn|$$

Вариант 3. РАСПРЕДЕЛЕНИЕ РЭЛЕЯ

$$p(x) = \frac{x}{a1^2} \exp(-x^2 / (2a1^2))$$

$$a1 \in [0, \infty], \quad x \in [0, \infty]$$

$$F(x) = \int_{-\infty}^{\infty} p(x) dx = 1 - \exp(-x^2 / (2a1^2)) = \Phi(t), \quad t = x / a1$$

$$R = a1 |Rn|$$

Вариант 4. РАСПРЕДЕЛЕНИЕ МАКСВЕЛЛА

$$p(x) = \frac{2x^2}{a1^3 \sqrt{2\pi}} \exp(-x^2 / 2a1^2)$$

$$a1 \in [0, \infty], \quad x \in [0, \infty]$$

$$F(x) = \int_{-\infty}^{\infty} p(x) dx = \Phi(t) - 2t\varphi(t), \quad t = x / a1$$

$$\Phi(t) = 1 - \exp(-x^2 / (2a1^2)), \quad \varphi(t) = \frac{1}{\sqrt{2\pi}} \exp(-t^2 / 2)$$

$$R = a1 \sqrt{\sum_{i=1}^3 Rn_i^2}$$

Вариант 5. РАСПРЕДЕЛЕНИЕ ПАРЕТО

$$p(x) = a2^{a1} x^{-a1-1}$$

$$a1 \in [0, \infty], \quad a2 \in [0, \infty], \quad x \in [a2, \infty]$$

$F(x)$ определить прямым интегрированием

$$R = \frac{a2}{(1-U)^{1/a1}}$$

Вариант 6. РАСПРЕДЕЛЕНИЕ ЭРЛАНГА

$$p(x) = (x - a_2)^{m-1} \exp(-a_1(x - a_2))$$

$$a_1 \in [0, \infty], \quad a_2 \in [0, \infty], \quad x \in [a_2, \infty]$$

m — целое число

$F(x)$ определить прямым интегрированием

$$R = a_2 - \frac{1}{a_1} \ln \left(\prod_{i=1}^m U_i \right)$$

Вариант 7. РАСПРЕДЕЛЕНИЕ ЛАПЛАСА

$$p(x) = \frac{a_1}{2} \exp(-a_1 \cdot |x - a_2|)$$

$$a_1 \in [0, \infty], \quad a_2 \in [-\infty, \infty], \quad x \in [-\infty, \infty]$$

$F(x)$ определить прямым интегрированием

$$R = \begin{cases} a_2 + \frac{1}{a_1} \ln(2U), & U < 0.5 \\ a_2 - \frac{1}{a_1} \ln(2(1-U)), & U > 0.5 \\ a_2, & U = 0 \end{cases}$$

Вариант 8. НОРМАЛЬНОЕ РАСПРЕДЕЛЕНИЕ

$$p(x) = \frac{1}{a_2 \sqrt{2\pi}} \exp(-(x - a_1)^2 / (2a_2^2))$$

$$a_1 \in [-\infty, \infty], \quad a_2 \in [0, \infty], \quad x \in [-\infty, \infty]$$

$F(x)$ определить прямым интегрированием

$$R = \frac{\sqrt{-2 \ln(U_1)} \cdot \sin(2\pi U_2)}{\sqrt{-2 \ln(U_1)} \cdot \cos(2\pi U_2)}$$

$$Rn = a_1 + a_2 R$$

Вариант 9. ЛОГНОРМАЛЬНОЕ РАСПРЕДЕЛЕНИЕ

$$p(x) = \frac{1}{x a_2 \sqrt{2\pi}} \exp(-(\ln(x) - a_1)^2 / (2a_2^2))$$

$$a_1 \in [-\infty, \infty], \quad a_2 \in [0, \infty], \quad x \in [0, \infty]$$

$F(x)$ определить прямым интегрированием

$$R = \frac{\sqrt{-2 \ln(U_1)} \cdot \sin(2\pi U_2)}{\sqrt{-2 \ln(U_1)} \cdot \cos(2\pi U_2)}$$

$$Rn = \exp(a1 + a2R)$$

Вариант 10. РАСПРЕДЕЛЕНИЕ КОШИ

$$p(x) = \frac{a1}{\pi(a1^2 + (x - a2)^2)}$$

$$a1 \in [0, \infty], \quad a2 \in [-\infty, \infty], \quad x \in [-\infty, \infty]$$

$F(x)$ определить прямым интегрированием

$$R = a2 + a1 * \operatorname{tg}(\pi(U - 0.5))$$

Вариант 11. ЛОГИСТИЧЕСКОЕ РАСПРЕДЕЛЕНИЕ

$$p(x) = \frac{1}{a2} \cdot \frac{\exp\left\{-\frac{(x - a1)}{a2}\right\}}{\left[1 + \exp\left\{-\frac{(x - a1)}{a2}\right\}\right]^2}$$

$$a1 \in [-\infty, \infty], \quad a2 \in [0, \infty], \quad x \in [-\infty, \infty]$$

$F(x)$ определить прямым интегрированием

$$R = a1 + a2 * \ln\left[\frac{1 - U}{U}\right]$$

Вариант 12. РАСПРЕДЕЛЕНИЕ ВЕЙБУЛЛА

$$p(x) = \frac{(x - a3)^{a1-1}}{a2^{a1}} \cdot \exp\left\{-\left(\frac{(x - a3)}{a2}\right)^{a1}\right\}$$

$$a1 \in [0, \infty], \quad a2 \in [0, \infty], \quad a3 \in [0, \infty], \quad x \in [a2, \infty]$$

$F(x)$ определить прямым интегрированием

$$R = a3 + a2 * [-\ln(U)]^{1/a1}$$

Вариант 13. РАСПРЕДЕЛЕНИЕ МИНИМАЛЬНОГО ЗНАЧЕНИЯ

$$p(x) = \frac{1}{a2} \exp\left(\frac{(x-a1)}{a2} - \exp\left\{-\frac{(x-a1)}{a2}\right\}\right)$$

$$a1 \in [-\infty, \infty], \quad a2 \in [0, \infty], \quad x \in [-\infty, \infty]$$

$F(x)$ определить прямым интегрированием

$$R = a1 + a2 * \ln[-\ln(1-U)]$$

Вариант 14. ГАММА РАСПРЕДЕЛЕНИЕ

$$p(x) = a2^{a1} (x-a3)^{a1-1} \exp(-a2(x-a3))$$

$$a1 \in [0, \infty], \quad a2 \in [0, \infty], \quad a3 \in [-\infty, \infty], \quad x \in [-\infty, \infty]$$

$F(x)$ определить прямым интегрированием

$$R = a3 + \frac{1}{a2} * [(q-1) \ln(U^{1/(1-c)})]$$

$$q = U^{1/a1} / (U^{1/a1} + U^{2/a1})$$

$$c = a1 - [a1]$$

[] — целая часть